




PCW

GAMES COLLECTION

FOR THE

AMSTRAD



PCW

GAMES
COLLECTION
FOR THE
AMSTRAD

PCW

GAMES COLLECTION

**FOR THE
AMSTRAD**



Illustrations by Patrick Lynn

Copyright ©Editors of PCW and Timothy Banks 1985

All rights reserved

First published in Great Britain in 1985 by

Century Communications Ltd,

12-13 Greek Street,

London W1V 5LE

ISBN 0 7126 0652 1

Printed in Great Britain by

Billings & Sons Ltd, Worcester

CONTENTS

<i>Introduction</i>	1
<i>Typing Tips</i>	2
Moonlander	4
Battleships	9
Euler's Tour	16
Fighter Pilot	22
Alien Blaster	27
Minesweeper	31
Zombies	37
Rocket Attack	42
Escape	47
Patience	51
Asteroids	61
Bomber	66
Cheese	70
Grand Prix	74
Minefield	78
Splash!	82
Pontoon	89
Target Practice	97
Duckshoot	102
Virus	107

INTRODUCTION

Each year *Personal Computer World*, Britain's biggest micro-computer magazine, receives literally thousands of computer programs from its dedicated band of readers. Now, for the first time, we have selected the very best of their collective efforts and proudly offer you this balanced compilation of twenty games. Within these pages you are sure to find something of pleasure, whether your taste is for arcade-action or board games, adventures or brain-teasers.

Each program has been redesigned by an Amstrad expert to make full use of colour, sound and the other facilities of your machine. All programs have been closely vetted by one of our experienced team of referees to ensure that they are as bug-free as possible.

Full documentation is included so that you can gain a clear understanding of how each program works. You will be able to improve your own programming style, and many of the routines can also be used within your own programs. Who knows, if you spend as many hours in front of your computer as these programmers have, you may get one of your programs published next year!

TYPING TIPS

If you are a new computer user, before you start typing in any programs it is important to read through your user manual so that you are familiar with the keyboard of your machine. Most of all, it is vital that you type in the program *exactly* as it appears in the book. To help you avoid the commonest mistakes, here are some hints.

- 1 Think before you type. Try to understand as much of what the program is trying to do as you can.
- 2 Remember that every character is important to your computer and you should type in exactly what is listed in the book, including all spaces. It is not, however, necessary to have any spaces between the line number and the beginning of the text.
- 3 Be particularly careful that you distinguish between the following:
 - 1 (one) and I (letter)
 - 0 (zero and O (letter). All zeros are Ø in the listings
 - : (colon) and ; (semi-colon)
 - , (comma) and . (full stop)
 - " (quote marks) and ' (apostrophe)
 - | and : (colon)
 - " " "" quote-marks with without a space between them.
- 4 When lower case letters are listed they must be entered as such, not as capitals.
- 5 Always save your program on a tape before running it for the first time. If you have made any typing errors your programs may crash and hours of patient work be lost before you can correct it.
- 6 You may come across the symbol in the listings. This is, in fact, the character found on the Amstrad by pressing **SHIFT** and the **P** key simultaneously.

Note:

Turn your computer off, and then on again, between games, to ensure that the memory is completely cleared out each time.

MOONLANDER

In this game your objective is to land a succession of lunar modules on a landing platform which is in constant motion along the surface of the moon. There are ten levels of difficulty, with ten being the easiest and one the most difficult.

The game will end after you crash five modules. You may move the module to either the left or the right by using the joystick or keyboard controls; key L moves it to the left, key O to the right.

You score ten points for landing the module in the centre of the platform and five points for landing it on the side of the platform.

Detailed description

25 Change mode.

30-70 Install the user-defined graphics characters.

80-130 Set up the ink colours, border, paper and pen.

140-220 Print the instructions on the screen.

230-250 Set the variables to their initial values and clear the buffer.

260-270 Ask the player to enter the level of difficulty required and check that the value is within the permitted range.

280-370 Set the variable for the length of the sound loop to the appropriate figure depending on the level of difficulty selected.

380-410 Clear the screen and print a starry background at random on the screen.

420-440 Print the moonscape at random on the bottom line.

450 Set a string variable for the moonlander vehicle.

460 Set a string variable for the landing platform.

470 Generate a random number for the positioning of the landing platform.

480 Set the direction counter to one.
 490 Sound loop for the descent of the moonlander.
 500 Start of loop for the descent of the moonlander.
 510 Calculation of random moonscape immediately adjacent to the landing platform.
 520 Increment the landing platform position counter.
 530-540 Check to ensure that the platform has not moved off the screen.
 550 Delete the moonlander.
 560-570 Check the keyboard for entries and make the appropriate changes to the variable x which controls the horizontal position of the moonlander.
 580 Print the landing platform.
 590 Print the moonlander.
 600 Delay loop; the length of this depends on the level of difficulty selected.
 610-620 Print score and lives.
 630 Check for landings made off the platform.
 640-680 Check for landings made on the platform; update score and play a tune.
 690-740 Print exploded moonlander and initiate sound of the crash.
 750-770 Decrement lives. If lives are zero then go to the end of game routine.
 780-820 Print end of game message and play a tune.
 Invite the player to have another game.

```

10 REM MOON LANDER
20 REM T.BANKS
25 MODE 1
30 SYMBOL AFTER 249
40 SYMBOL 255,&18,&3C,&7E,&FF,&FF,&C3,&8
  1,&81
50 SYMBOL 254,&0,&0,&0,&0,&0,&FF,&C0,&C0
:SYMBOL 253,&0,&0,&0,&0,&0,&FF,&0,&0:SYM
BOL 252,&0,&0,&0,&0,&0,&FF,&3,&3
60 SYMBOL 251,&0,&0,&0,&0,&20,&74,&FF,&F
  F
70 SYMBOL 250,&0,&20,&30,&38,&3C,&7C,&FF
  ,&FF
  
```

```

80 INK 0,0
90 INK 1,26
100 INK 2,6
110 INK 3,15
120 BORDER 0
130 PAPER 0:PEN 1
140 CLS:LOCATE 15,1:PRINT "MOON LANDER":
PRINT:PRINT"IN THIS GAME YOUR OBJECTIVE
IS TO LAND A SUCCESSION OF LUNAR MODULE
S ON THE LANDING PLATFORM WHICH IS IN
CONSTANT MOTION ALONG THE SURFACE OF
THE MOON."
150 PRINT:PRINT"THERE ARE 10 LEVELS OF D
IFFICULTY, WITH 10 BEING THE EASIEST.":P
RINT:PRINT"THE GAME WILL END AFTER YOU H
AVE CRASHEDFIVE MODULES."
160 PRINT:PRINT"YOU SCORE 10 POINTS FOR
LANDING THE MODULE IN THE CENTRE OF
THE PLATFORM AND5 POINTS FOR LANDING IT
OFF CENTRE."
170 LOCATE 5,25:PRINT "PRESS <ENTER> TO
CONTINUE"
180 IF INKEY(18)<>0 THEN GOTO 180
190 CLS:LOCATE 15,1:PRINT "MOON LANDER"
200 PRINT:PRINT"CONTROLS ":"PRINT:PRINT"
KEYBOARD : 1 MOVES MODULE LEFT
0 MOVES MODULE RIGHT":PRINT:P
RINT"JOYSTICK MAY ALSO BE USED."
210 LOCATE 5,25:PRINT "PRESS <ENTER> TO
START"
220 IF INKEY(18)<>0 THEN GOTO 220
230 LET X=14
240 LET LIVES=5:LET SCORE=0
250 FOR GF=1 TO 50:BUFFER$=INKEY$:NEXT
260 PRINT CHR$(13):CLS:INPUT "LEVEL OF
DIFFICULTY (1-10) ";L
270 IF L<1 OR L>10 THEN GOTO 260
280 IF L=1 THEN LET T=410
290 IF L=2 THEN LET T=450
300 IF L=3 THEN LET T=490
310 IF L=4 THEN LET T=540
320 IF L=5 THEN LET T=600
330 IF L=6 THEN LET T=650
340 IF L=7 THEN LET T=710
350 IF L=8 THEN LET T=760

```



```

360 IF L=9 THEN LET T=800
370 IF L=10 THEN LET T=850
380 CLS
390 FOR A=1 TO 200
400 PLOT INT(RND*640)+1,INT(RND*400)+1
410 NEXT
420 FOR W=1 TO 40
430 IF INT(RND*2)+1=1 THEN LOCATE W,25:P
PRINT CHR$(251) ELSE LOCATE W,25:PRINT CH
R$(250)
440 NEXT
450 LET A$=CHR$(255)
460 LET B$=CHR$(251)+CHR$(254)+CHR$(253)
+CHR$(252)+CHR$(250)
470 LET F=INT(RND*34)+1
480 LET G=1
490 SOUND 1,0,T,7,0,0,31
500 FOR C=2 TO 24
510 IF INT(RND*2)+1=1 THEN LET B$=CHR$(2
51)+CHR$(254)+CHR$(253)+CHR$(252)+CHR$(2
50) ELSE LET B$=CHR$(250)+CHR$(254)+CHR$
(253)+CHR$(252)+CHR$(251)
520 LET F=F+G
530 IF F>30 THEN LET G=-1
540 IF F=1 THEN LET G=1
550 LOCATE X,C-1:PEN 0:PRINT " ":PEN 1
560 IF INKEY(64)=0 OR JOY(0)=4 THEN LET
X=X-1:IF X<1 THEN LET X=1
570 IF INKEY(32)=0 OR JOY(0)=8 THEN LET
X=X+1:IF X>40 THEN LET X=40
580 LOCATE F,24:PRINT B$
590 LOCATE X,C:PEN 3:PRINT A$:PEN 1
600 FOR Y=1 TO (2*L)*10:NEXT
610 LOCATE 1,1:PEN 2:PRINT "SCORE ";SCOR
E:LOCATE 25,1:PRINT "LIVES ";LIVES:PEN 1
620 NEXT
630 IF X<>F+3 AND X<>F+1 AND X<> F+2 THE
N GOTO 690
640 IF X=F+2 THEN LET SCORE=SCORE+10 ELS
E LET SCORE=SCORE+5
650 ENT 1,15,-20,5
660 SOUND 1,300,0,15,15,1
670 FOR Z=1 TO 1000:NEXT
680 GOTO 380
690 LOCATE X,24:PEN 2:PRINT CHR$(238):PE
N 1

```

```
700 ENV 1,1,0,50,14,-1,10
710 SOUND 1,0,0,15,1,0,31
720 SOUND 2,0,0,15,1,0,15
730 SOUND 4,0,0,15,1,0,1
740 FOR U=1 TO 1600:NEXT
750 LET LIVES=LIVES-1
760 IF LIVES=0 THEN GOTO 780
770 GOTO 380
780 CLS:PRINT "BAD LUCK YOU CRASHED ALL
OF YOUR SHIPS":PRINT:PRINT"YOU OBTAINED
A SCORE OF ";SCORE:PRINT:PRINT"ON DIFFIC
ULTY LEVEL ";L
790 ENT 2,30,20,5:SOUND 4,300,0,15,15,2
800 LOCATE 5,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
810 IF INKEY(18)<>0 THEN GOTO 810
820 RUN 230
```

BATTLESHIPS



This is a computer version of the traditional pen and pencil game of battleships, where you pit your wits against the Amstrad. The object of the game is to sink the computer's fleet, before it sinks yours.

The game is played on a ten-by-ten grid upon which you have to place your fleet of six ships. To do this simply type in the grid reference letter and number, then press the large ENTER key. Submarines occupy one square each, destroyers and frigates two, and the cruiser takes up three squares.

The computer's fleet is shown on the right-hand side of the screen and yours is on the left. To fire at an opposing craft, type in its grid reference; but you need to fire as many shots as the craft takes up squares, so it's one for a sub, two for frigates and destroyers, and three for a cruiser. But be careful not to shoot into a square you have already hit.

The computer then takes its turn to fire at your fleet and the game continues in this manner until a victory is announced.

Detailed description

30-120 Install the user-defined graphics characters.

130-150 Set the computer to mode 1. Set the ink, border and paper colours. Dimension the arrays for both fleets.

160-210 Clear the screen and print the instructions for playing the game.

220-310 Clear the buffer. Clear the screen and print two ten by ten grids.

320 Dimension the fleet arrays for both the player and the computer to position their ships.

330-640 Invite the player to position a ship on the grid and check that the square is not already occupied. Then repeat the procedure for each ship.

650-760 Place the computer's ships into the array.

770-910 Invite the player to select a square to shell. A check is then made to see if a hit has been registered, and if so a letter is placed in the appropriate square to indicate the type of ship which has been hit. At the same time a sound is generated to indicate a hit has been scored. A further check is then made to see if all the ships have been sunk and if so the program goes to the end of game routine.

920-980 Select at random a position on the player's grid for the computer to shell. If a hit is scored then the score counter is incremented and a check is made to see if the score has reached eleven which means that all the ships have been sunk. If so, go to the end of game routine. If a hit has been scored and it is not a submarine (which occupies only one square), then the flag (hit) is set so that on subsequent moves the computer will search either side of the hit to destroy the rest of the ship.

990-1090 Check that the player's input is valid.

1100 Clear the text at the bottom of the screen.

1110-1170 Check the input position of each ship to ensure it does not run off the side of the grid.

1180-1220 If the computer has won print a short message and play a tune. Invite the player to try again.

1230-1250 If the player has won, print a congratulations message, play a tune, and ask the player if another game is wanted.

10 REM BATTLESHIPS

20 REM T.BANKS

30 SYMBOL AFTER 240

40 SYMBOL 255,&0,&0,&0,&0,&10,&7E,&0,&0

50 SYMBOL 254,&0,&0,&0,&1,&6D,&3F,&0,&0

60 SYMBOL 253,&0,&0,&0,&20,&20,&FC,&0,&0


```

70 SYMBOL 252,&0,&2,&2,&F,&7F,&3F,&0,&0
80 SYMBOL 251,&0,&0,&0,&B0,&FC,&FC,&0,&0
90 SYMBOL 250,&0,&0,&1,&19,&7F,&3F,&0,&0
100 SYMBOL 249,&0,&CC,&CC,&FD,&FF,&FF,&0
    ,&0
110 SYMBOL 248,&0,&0,&0,&98,&FC,&FC,&0,&
    0
120 SYMBOL 247,&FF,&81,&81,&81,&81,&81,&
    81,&FF
130 MODE 1
140 INK 0,1:INK 1,26:INK 2,18:INK 3,15:B
    ORDER 1:PAPER 0
150 DIM YOU2(10,10):DIM ENEMY2(10,10)
160 CLS:LOCATE 15,1:PRINT "BATTLESHIPS":
    PRINT:PRINT"THIS IS THE TRADITIONAL GAME
    OF          BATTLESHIPS IN WHICH YOU PIT
    YOUR WITS  AGAINST THE AMSTRAD."
170 PRINT:PRINT"EACH FLEET COMPRISES TWO
    SUBMARINES,      TWO DESTROYERS, A FRIGAT
    E AND A CRUISER WHICH ARE ENTERED ON A T
    EN BY TEN GRID."
180 PRINT:PRINT"A SUBMARINE OCCUPIES ONE
    SQUARE, A        DESTROYER OR FRIGATE TWO
    SQUARES AND THECRUISER THREE SQUARES."
190 PRINT:PRINT"ALL OCCUPIED SQUARES MUS
    T BE HIT BEFORE A SHIP IS SUNK."
191 PRINT
192 PRINT "YOUR FLEET WILL BE DISPLAYED
    ON THE LEFTHAND SIDE OF THE SCREEN."
194 PRINT:PRINT "A HIT ON THE ENEMY IS R
    ECORDED BY A     LETTER DENOTING THE TYP
    E OF SHIP YOU    HAVE HIT."
200 LOCATE 5,25:PRINT "PRESS <ENTER> TO
    START"
210 IF INKEY(18)<>0 THEN GOTO 210
220 FOR Q=1 TO 10:LET BUFFER$=INKEY$:NEX
    T
230 CLS
240 LOCATE 15,1:PRINT "BATTLESHIPS"
250 LET A$=CHR$(247)+CHR$(247)+CHR$(247)
    +CHR$(247)+CHR$(247)+CHR$(247)+CHR$(247)
    +CHR$(247)+CHR$(247)+CHR$(247)
260 FOR A=5 TO 14
270 PEN 1

```

```

280 LOCATE 5,A:PRINT A-5;A$;"      ";A-5;A
$
290 PRINT
300 NEXT
310 LOCATE 8,4:PRINT "ABCDEFGH IJ";"
";"ABCDEFGH IJ"
320 DIM ENEMY(10,10):DIM YOU(10,10)
330 LOCATE 2,18:PRINT "PLEASE TYPE IN TH
E POSITION OF YOUR      FLEET (LETTER,NUM
BER)"
340 LOCATE 1,24:PRINT "THEN PRESS <ENTER
>
350 PEN 2
360 GOSUB 1100:LOCATE 10,22:INPUT "SUBMA
RINE 1 ";J$:GOSUB 1040
370 IF FLAG=1 THEN LET FLAG=0:GOTO 360
380 IF YOU(X,Y)=1 OR YOU(X,Y)=2 OR YOU(X
,Y)=3 OR YOU(X,Y)=4 THEN GOTO 360
390 LOCATE 7+X,4+Y:PRINT CHR$(255):LET Y
OU(X,Y)=1
400 GOSUB 1100:LOCATE 10,22:INPUT "SUBMA
RINE 2 ";J$:GOSUB 1040
410 IF FLAG=1 THEN LET FLAG=0:GOTO 400
420 IF YOU(X,Y)=1 OR YOU(X,Y)=2 OR YOU(X
,Y)=3 OR YOU(X,Y)=4 THEN GOTO 400
430 LOCATE 7+X,4+Y:PRINT CHR$(255):LET Y
OU(X,Y)=1
440 GOSUB 1100:LOCATE 10,22:INPUT"DESTRO
YER 1";J$:GOSUB 1040
450 IF FLAG=1 THEN LET FLAG=0:GOTO 440
460 IF X=10 THEN GOTO 440
470 IF YOU(X,Y)>0 OR YOU(X+1,Y)>0 THEN G
OTO 440
480 LOCATE 7+X,4+Y:PRINT CHR$(254);CHR$(
253):LET YOU(X,Y)=2:LET YOU(X+1,Y)=2
490 GOSUB 1100:LOCATE 10,22:INPUT"DESTRO
YER 2";J$:GOSUB 1040
500 IF FLAG=1 THEN LET FLAG=0:GOTO 490
510 IF X=10 THEN GOTO 490
520 IF YOU(X,Y)>0 OR YOU(X+1,Y)>0 THEN G
OTO 490
530 LOCATE 7+X,4+Y:PRINT CHR$(254);CHR$(
253):LET YOU(X,Y)=2:LET YOU(X+1,Y)=2
540 GOSUB 1100:LOCATE 10,22:INPUT "FRIGA
TE ";J$:GOSUB 1040

```

```

550 IF X=10 THEN GOTO 540
560 IF FLAG=1 THEN LET FLAG=0:GOTO 540
570 IF YOU(X,Y)>0 OR YOU(X+1,Y)>0 THEN G
OTO 540
580 LOCATE 7+X,4+Y:PRINT CHR$(252);CHR$(
251):LET YOU(X,Y)=3:LET YOU(X+1,Y)=3
590 GOSUB 1100:LOCATE 10,22:INPUT "CRUIS
ER ";J$:GOSUB 1040
600 IF X=10 OR X=9 THEN GOTO 590
610 IF FLAG=1 THEN LET FLAG=0:GOTO 590
620 IF YOU(X,Y)>0 OR YOU(X+1,Y)>0 OR YOU
(X+2,Y)>0 THEN GOTO 590
630 LOCATE 7+X,4+Y:PRINT CHR$(250);CHR$(
249);CHR$(248):LET YOU(X,Y)=4:LET YOU(X+
1,Y)=4:LET YOU(X+2,Y)=4
640 PEN 1
650 REM ENEMY POSITION
660 LET ENEMY(INT(RND*10)+1,INT(RND*10)+
1)=1
670 LET R=10:GOSUB 1110
680 LET ENEMY(0,P)=1
690 LET R=9:GOSUB 1110
700 LET ENEMY(0,P)=2:LET ENEMY(0+1,P)=2
710 LET R=9:GOSUB 1110
720 LET ENEMY(0,P)=2:LET ENEMY(0+1,P)=2
730 LET R=9:GOSUB 1110
740 LET ENEMY(0,P)=3:LET ENEMY(0+1,P)=3
750 LET R=8:GOSUB 1110
760 LET ENEMY(0,P)=4:LET ENEMY(0+1,P)=4:
LET ENEMY(0+2,P)=4
770 LOCATE 1,19:PRINT "
780 LOCATE 1,20:INPUT "POSITION TO BOMB
";J$:GOSUB 1040
790 IF FLAG=1 THEN GOSUB 1100:LET FLAG=0
:GOTO 780
800 IF YOU2(X,Y)=1 THEN GOTO 780
810 IF ENEMY(X,Y)=0 THEN LET YOU2(X,Y)=1
:LOCATE X+24,4+Y:PEN 3:PRINT "*":PEN 1:G
OTO 890
820 PEN 3:ENV 1,1,0,50,14,-1,10:SOUND 1,
0,0,15,1,0,31:SOUND 2,0,0,15,1,0,15:SOUN
D 4,0,0,15,1,0,1
830 FOR LOOP = 1 TO 100:NEXT
840 IF ENEMY(X,Y)=1 THEN LET YOU2(X,Y)=1
:LOCATE X+24,4+Y:LET HCOUNT=HCOUNT+1:PRI

```

```

NT "S":GOTO 890
850 IF ENEMY(X,Y)=2 THEN LET YOU2(X,Y)=1
:LOCATE X+24,4+Y:LET HCOUNT=HCOUNT+1:PRI
NT "D":GOTO 890
860 IF ENEMY(X,Y)=3 THEN LET YOU2(X,Y)=1
:LOCATE X+24,4+Y:LET HCOUNT=HCOUNT+1:PRI
NT "F":GOTO 890
870 IF ENEMY(X,Y)=4 THEN LET YOU2(X,Y)=1
:LOCATE X+24,4+Y:LET HCOUNT=HCOUNT+1:PRI
NT "C":GOTO 890
880 END
890 FOR LOOP2=1 TO 600:NEXT:PEN 1: LOCAT
E 1,20:PRINT "COMPUTER'S MOVE  "
900 PEN 1:IF HCOUNT=11 THEN GOTO 1230
910 IF HIT =1 THEN GOTO 990
920 LET X1=INT(RND*10)+1:LET Y1=INT(RND*
10)+1:IF ENEMY2(X1,Y1)=1 THEN GOTO 920
930 IF YOU(X1,Y1)=1 THEN LET COUNT=COUNT
+1
940 IF YOU(X1,Y1)=2 THEN LET HIT=1:LET D
S=1:LET COUNT=COUNT+1
950 IF YOU(X1,Y1)=3 THEN LET HIT=1:LET F
G=1:LET COUNT=COUNT+1
960 IF YOU(X1,Y1)=4 THEN LET HIT=1:LET C
R=1:LET COUNT=COUNT+1
970 IF COUNT=11 THEN GOTO 1180
980 LET ENEMY2(X1,Y1)=1 :FOR T1=1 TO 100
:LOCATE X1+7,Y1+4:PRINT "*":LOCATE X1+7,
Y1+4:PRINT " ":NEXT:LOCATE X1+7,Y1+4:PRI
NT "*":GOTO 770
990 IF X1<10 AND DS=1 AND DS1=0 THEN LET
X1=X1+1:IF YOU(X1,Y1)=2 THEN LET COUNT=
COUNT+1:LET HIT=0:LET DS=0:GOTO 970:ELSE
LET DS1=1
1000 IF DS=1 AND DS1=1 THEN LET X1=X1-2:
IF YOU(X1,Y1)=2 THEN LET COUNT=COUNT+1:L
ET HIT=0:LET DS=0:GOTO 970
1010 IF X1<10 AND FG=1 AND FG1=0 THEN LE
T X1=X1+1:IF YOU(X1,Y1)=3 THEN LET COUNT
=COUNT+1:LET HIT=0:LET FG=2:GOTO 970:ELS
E LET FG1=1
1020 IF FG=1 AND FG1=1 THEN LET X1=X1-2:
IF YOU(X1,Y1)=3 THEN LET COUNT=COUNT+1:
LET HIT=0:LET FG=2:GOTO 970
1030 LET HIT=0:GOTO 970

```

```

1040 LET J$=UPPER$(J$)
1050 IF J$="" THEN LET FLAG=1:RETURN
1060 IF LEN (J$)>2 THEN LET FLAG=1:RETURN
1070 LET X=ASC(LEFT$(J$,1)):LET X=X-64:IF X<1 OR X>10 THEN LET FLAG=1:RETURN
1080 LET Y=ASC(RIGHT$(J$,1)):LET Y=Y-47:IF Y<1 OR Y>10 THEN LET FLAG=1:RETURN
1090 RETURN
1100 LOCATE 1,22:PRINT "
":RETURN
1110 LET O=INT(RND*R)+1:LET P=INT(RND*10)+1
1120 IF ENEMY(O,P)>0 THEN GOTO 1110
1130 IF R=10 THEN RETURN
1140 IF R=9 AND ENEMY(O+1,P)>0 THEN GOTO 1110
1150 IF R=2 THEN RETURN
1160 IF R=8 AND ( ENEMY(O+1,P)>0 OR ENEMY(O+2,P)>0) THEN GOTO 1110
1170 RETURN
1180 CLS:PRINT "I AM SORRY BUT THE COMPUTER HAS BEATEN YOU"
1190 ENT 1,30,20,5:SOUND 3,300,0,15,15,1
1200 LOCATE 5,25:PRINT "PRESS <ENTER> TO PLAY AGAIN"
1210 IF INKEY(18)<>0 THEN GOTO 1210
1220 RUN
1230 CLS:PRINT "CONGRATULATIONS!":PRINT:PRINT"YOU HAVE BEATEN THE AMSTRAD"
1240 ENT 2,30,-20,15:SOUND 3,300,0,15,15,2
1250 GOTO 1200

```

EULER'S TOUR

In this game of logic you will need all your powers of concentration to achieve the maximum score. The object of the game is to move a chess Knight around an eight-by-eight board visiting each square only once. The Knight moves in the traditional manner, that is two squares followed by one square to either side. The demonstration program will show this more clearly to those of you not familiar with chess.

As each square is visited it will be blanked out. If you become trapped you may move into a square previously visited, but if such a move is made then a warning buzzer will sound and your final score (expressed as a percentage) will be reduced. To move the Knight you just type in the X and Y co-ordinates of the square you wish to visit on the tour, pressing the large ENTER key after each one.

At the beginning of the game you will be asked if you wish the computer to demonstrate a perfect tour. You will be asked to choose the X and Y co-ordinates for the start of the tour. The Knight will take up this position and then move round the board, but you have to press the large ENTER key before each one. It should take you quite some time and a number of attempts to achieve the maximum score of 100% yourself. Good Luck!

Detailed description

40-60 Set the mode and the pen, ink and border colours.

70 Go to the sub-routine for printing the instructions.

80-210 Install the user-defined graphics characters.

220-230 Dimension the array for the board.

240-250 Restore the data pointer and set the variables to their initial values.

260-280 Read the data into the board array.

290 Dimension the used squares array.

300-370 Data for a perfect solution to Euler's Tour.

380-420 Set up the strings A\$ and B\$ to represent the

board, then clear the screen.

430-490 Print the board on the screen.

500-530 Ask the player if a demonstration is required.

540-610 Request the player to enter the X and Y coordinates for the initial position of the Knight, and check that the entry is valid. If it is not, repeat the request.

620-710 Routine to demonstrate a perfect tour. The movement of the Knight is controlled by the variable C% and uses data from the array BOD%.

720-750 Print a message and invite the player to try again.

760 Print the Knight at the specified location.

770-870 Print the instructions on the screen.

880-1080 Ask the player to enter the X and Y coordinates for the Knight's move and check that they are valid. The move is then made and the number of correct moves updated. If the square has not been previously visited then it is blanked out.

1090-1140 End of game routine where the score is printed and the player is invited to have another game.

```
10 REM EULERS TOUR
20 REM BY ALAN HUGHES
30 REM ADAPTED BY T.BANKS
40 MODE 1
50 PEN 1
60 INK 0,0: BORDER 0
70 GOSUB 770
80 REM USER DEFINED GRAPHICS
90 SYMBOL AFTER 240
100 SYMBOL 255,&FF,&80,&80,&80,&80,&80,&80,&80,&80
110 SYMBOL 254,&FF,&1,&1,&1,&1,&1,&1,&1,&1
120 SYMBOL 253,&80,&80,&80,&80,&80,&80,&80,&80,&FF
130 SYMBOL 252,&1,&1,&1,&1,&1,&1,&1,&1,&FF
140 SYMBOL 251,&0,&0,&1,&3,&7,&F,&1E,&14
150 SYMBOL 250,&0,&A0,&F0,&F8,&F8,&F8,&F8,&F8
160 SYMBOL 249,&1,&3,&3,&3,&3,&7,&7,&0
170 SYMBOL 248,&F8,&F8,&F0,&F0,&F0,&F8,&F8,&0
```

```

180 SYMBOL 247,&0,&7F,&7F,&7F,&7F,&7F,&7F,&7F
190 SYMBOL 246,&0,&FE,&FE,&FE,&FE,&FE,&FE,&FE
200 SYMBOL 245,&7F,&7F,&7F,&7F,&7F,&7F,&7F,&0
210 SYMBOL 244,&FE,&FE,&FE,&FE,&FE,&FE,&FE,&0
220 REM INITIALISE
230 DIM BOD%(63)
240 RESTORE
250 LET SCORE=0:LET WRONG=0
260 FOR A%=0 TO 63
270 READ BOD%(A%)
280 NEXT
290 DIM USED%(7,7)
300 DATA 17,16,8,20,10,11,12,22
310 DATA 25,3,0,1,2,7,4,5
320 DATA 26,27,24,29,30,6,28,13
330 DATA 9,19,32,21,18,23,15,14
340 DATA 49,48,40,45,42,31,44,54
350 DATA 50,35,57,33,34,39,36,37
360 DATA 58,59,56,61,62,63,60,38
370 DATA 41,51,52,53,43,55,47,46
380 FOR B=1 TO 8
390 LET A$=A$+CHR$(255)+CHR$(254)
400 LET B$=B$+CHR$(253)+CHR$(252)
410 NEXT
420 CLS
430 REM DRAW BOARD
440 FOR C=0 TO 14 STEP 2
450 LOCATE 12,C+6:PRINT A$:LOCATE 12,C+7
:PRINT B$
460 NEXT
470 LOCATE 9,4:PRINT "    1 2 3 4 5 6 7 8
"
480 FOR U=7 TO 21 STEP 2:LOCATE 28,U:LET
I=I+1:PRINT I:NEXT
490 LOCATE 20,3:PRINT "X":LOCATE 30,13:P
RINT "Y":LOCATE 15,1:PRINT "EULER'S TOUR
"
500 REM INPUT STARTING SQUARE
510 LOCATE 1,25:INPUT "COMPUTER DEMONSTR
ATION (Y/N) ";H$
520 IF UPPER$(H$)<>"Y" AND UPPER$(H$)<>"

```



```

N" THEN GOTO 510
530 IF UPPER$(H$)="N" THEN GOTO 880
540 REM COMPUTER DEMONSTRATION
550 LOCATE 1,25:PRINT "
":LOCATE 1,25:INPUT
"X POSITION ",X%
560 IF X%<1 OR X%>8 THEN GOTO 550
570 LOCATE 1,25:PRINT "
"
580 LOCATE 1,25:INPUT "Y POSITION ",Y%
590 IF Y%<1 OR Y%>8 THEN GOTO 570
600 LOCATE 1,25:PRINT "
"
610 X%=X%-1:Y%=Y%-1
620 FOR M=1 TO 63
630 GOSUB 760
640 C%=BOD%((Y%*8)+x%)
650 LOCATE 1,25:PRINT "PRESS <ENTER> TO
MOVE":IF INKEY(18)<>0 THEN GOTO 650
660 LOCATE (X%*2)+12,(Y%*2)+6:PEN 3:PRIN
T CHR$(247);CHR$(246):LOCATE (X%*2)+12,(
Y%*2)+7:PRINT CHR$(245);CHR$(244):PEN 1
670 X%=C% MOD 8
680 Y%= INT(C%/8)
690 FOR T=1 TO 200:NEXT
700 NEXT
710 GOSUB 760
720 LOCATE 1,25:PRINT "
":LOCATE 16,23:PRIN
T "FINISHED":LOCATE 5,25:PRINT "PRESS <E
NTER> TO PLAY AGAIN"
730 IF INKEY(18)<>0 THEN GOTO 730
740 FOR G=1 TO 255:LET J$=INKEY$:NEXT
750 RUN
760 LOCATE (X%*2)+12,(Y%*2)+6:PRINT CHR$
(251);CHR$(250):LOCATE (X%*2)+12,(Y%*2)+
7:PRINT CHR$(249);CHR$(248):RETURN
770 REM INSTRUCTIONS
780 CLS
790 LOCATE 15,1:PRINT "EULER'S TOUR":PRI
NT
800 PRINT "THE IDEA OF THIS GAME IS TO V
ISIT EVERY SQUARE ON THE BOARD WITH A KN
IGHT, ONLY VISITING EACH SQUARE ONCE."
810 PRINT:PRINT "YOU CAN WATCH THE COMPU

```

```

TER SOLVE THE      PROBLEM OR SOLVE IT YOU
RSELF.":PRINT:PRINT "WITH EITHER OPTION
YOU MUST ENTER A      STARTING POINT IN T
HE FORM X CO-ORDINATE Y CO-ORDINATE."
820 PRINT:PRINT"WHEN ATTEMPTING THE TOUR
YOURSELF YOU      MUST ENTER THE CO-ORDINA
TES OF EACH      SUCCESSIVE MOVE.":PRINT
830 PRINT"YOUR SCORE IS DISPLAYED AT THE
END OF      THE GAME AS A PERCENTAGE OF CO
RRECT      MOVES. IF YOU GET INTO DIFFICU
LTY YOU      MAY VISIT THE SAME SQUARE MORE
THAN      ONCE, BUT THIS WILL REDUCE YOU
R SCORE."
840 LOCATE 5,25:PRINT "PRESS <ENTER> TO
START"
850 IF INKEY(18)<>0 THEN GOTO 850
860 FOR O=1 TO 255:LET O%=INKEY$:NEXT
870 RETURN
880 REM PLAYERS MOVE
890 LOCATE 1,25:PRINT "
"
900 WHILE X%<1 OR X%>8:LOCATE 1,25:INPUT
"X POSITION ",X%:WEND
910 WHILE Y%<1 OR Y%>8:LOCATE 1,25:INPUT
"Y POSITION ",Y%:WEND
920 LET X%=X%-1:LET Y%=Y%-1
930 GOSUB 760
940 USED%(X%,Y%)=USED%(X%,Y%)+1:CORRECT=
1
950 WHILE CORRECT <64
960 GOSUB 760
970 LET X=X%:LET Y=Y%
980 LOCATE 1,25:PRINT "
"
990 LOCATE 1,25:INPUT "X POSITION ",X%
1000 IF X%<1 OR X%> 8 THEN GOTO 990
1010 LOCATE 1,25:INPUT "Y POSITION ",Y%
1020 IF Y%<1 OR Y%>8 THEN GOTO 1010
1030 LET X%=X%-1:LET Y%=Y%-1
1040 IF (NOT((((X%=X+2) OR (X%=X-2)) AND
((Y%=Y-1) OR (Y%=Y+1))) OR (((X%=X+1) O
R (X%=X-1)) AND ((Y%=Y-2) OR (Y%=Y+2))))
) THEN GOTO 980
1050 USED%(X%,Y%)=USED%(X%,Y%)+1

```

```

1060 IF USED%(X%,Y%)>1 THEN WRONG=WRONG+
1:SCORE=SCORE-2:SOUND 1,300,5,7:ELSE SCO
RE=SCORE+2:CORRECT =CORRECT +1
1070 LOCATE (X*2)+12,(Y*2)+6:PEN 3:PRIN
T CHR$(247);CHR$(246):LOCATE (X*2)+12,(Y
*2)+7:PRINT CHR$(245);CHR$(244):PEN 1
1080 WEND
1090 LOCATE 1,25:PRINT "
"
1100 LOCATE 1,23:PRINT "PROBLEM SOLVED !
SCORE = ";SCORE/128*100; " %"
1110 LOCATE 5,25:PRINT"PRESS <ENTER> TO
PLAY AGAIN"
1120 IF INKEY(18)<>0 THEN GOTO 1120
1130 FOR P=1 TO 255:LET J%=INKEY$:NEXT
1140 RUN

```



This is a standard type of aeroplane game where you are the pilot of a fighter plane intent on shooting down the enemy bomber.

This is not easy as it is in constant random motion around the screen taking avoiding action. There is a time limit and when the clock reaches zero you run out of fuel and have to return to base.

You destroy the bomber by getting it in your gunsight and then firing - but you have to score fifty to shoot it down. Another bomber then takes its place. Your gunsight remains in the centre of the screen and you can use either joystick or the following keys on the numeric keypad to get the bomber in the sights:

8	Climb
2	Dive
4	Left
6	Right
SPACE	Fire

Detailed description

30-80 Set the ink, border, paper and pen colours.

90 Set the initial time.

100-160 Clear the screen. Print the instructions for playing the game.

170-180 Install the user-defined graphics characters.
190-200 Set the initial position of the bomber.
210-220 Clear the screen and print the machine guns.
230 Check that the time has not exceeded the limit of 250.
240 Print the gunsight.
250 Set the envelope to vary the volume of the sound.
260-340 Detect which key or joystick position has been selected and update the variables which control the position of the bomber. If the fire button or space bar has been pressed then fire the guns with the appropriate sound. A check is made to determine if the bomber is in the gunsight and if so the program goes to the score routine.
350 Move the position of the bomber at random.
360 Print the current score.
370-400 Check that the bomber has not gone off the screen. If it has then change the print co-ordinates to avoid this.
410-430 If the position of the bomber has changed then erase the image and reprint at the new position.
440 Register the new position of the bomber.
450-470 Create a short pause then update the time and print the time elapsed.
480 If the time has not elapsed then loop back to line 230 to continue the game.
490-520 If the player has run out of time then print a message and invite the player to try again.
530-550 Update and print the score when a hit has been registered. If the score exceeds fifty then go to the routine to explode the bomber.
560-570 Explode the bomber.
580-590 Generate the sound of an explosion and create a short delay.
600-630 Print a congratulations message and invite the player to have another game.

```
10 REM FIGHTER PILOT
20 REM T.BANKS
30 INK 0,6
40 INK 1,1
50 INK 2,24
```

```

60 INK 3,18
70 BORDER 1
80 PAPER 1:PEN 3
90 LET DATUM=((TIME/300))
100 CLS
110 LOCATE 10,7:PRINT "FIGHTER PILOT":PR
INT:PRINT"YOU ARE THE PILOT OF A FIGHTER
PLANE. YOUR TASK IS TO SHOOT DOWN THE
ENEMY BOMBER."
115 PRINT:PRINT "YOU HAVE TO SCORE 50 HI
TS TO DESTROY THEENEMY, BUT YOU ONLY HAV
E A LIMITED AMOUNT OF FUEL AND IF Y
OU RUN OUT YOU WILL NEED TO RETURN TO
BASE."
118 LOCATE 5,25:PRINT"PRESS <ENTER> TO C
ONTINUE"
119 IF INKEY(18)<>0 THEN GOTO 119
120 CLS:PRINT: PRINT: PRINT"CONTROLS :":
PRINT:PRINT"KEYBOARD: USING NUMERICAL DA
TA PAD":PRINT:PRINT " CLIMB":PRIN
T " 7 8 9 ":PRINT
130 PRINT"LEFT 4 6 RIGHT":PRINT:P
RINT " 1 2 3":PRINT " D
IVE"
140 PRINT: PRINT"PRESS <SPACE> TO FIRE":
PRINT:PRINT "ALTERNATIVELY, A JOYSTICK M
AY BE USED."
150 LOCATE 5,25:PRINT"PRESS <ENTER> TO S
TART"
160 IF INKEY(18)<>0 THEN GOTO 160
170 SYMBOL AFTER 240:SYMBOL 255,&0,&0,&0
,&0,&0,&18,&18,&18
180 SYMBOL 254,&0,&0,&0,&0,&C,&FF,&C,&0:
SYMBOL 253,&18,&18,&18,&3C,&7E,&FF,&7E,&
3C:SYMBOL 252,&0,&0,&0,&0,&30,&FF,&30,&0
190 LET X=20:LET Y=2
200 LET X1=20:LET Y1=2
210 CLS
220 LOCATE 8,25:PRINT "/":LOCATE 32,25:P
RINT "\"
230 WHILE TICKER < 250
240 LOCATE 20,13:PRINT "+"
250 ENV 1,1,0,5,1,-7,10,1,7,5,1,-7,10,1,
7,5
260 IF JOY(0)=1 OR INKEY(11)=0 THEN LET

```

```

Y=Y-1
270 IF JOY(0)=2 OR INKEY(14)=0 THEN LET
Y=Y+1
280 IF JOY(0)=4 OR INKEY(20)=0 THEN LET
X=X-1
290 IF JOY(0)=5 OR INKEY(10)=0 THEN LET
Y=Y-1:LET X=X-1
300 IF JOY(0)=9 OR INKEY(3)=0 THEN LET Y
=Y-1:LET X=X+1
310 IF JOY(0)=10 OR INKEY(5)=0 THEN LET
Y=Y+1:LET X=X+1
320 IF JOY(0)=6 OR INKEY(13)=0 THEN LET
Y=Y+1:LET X=X+1
330 IF JOY(0)=16 OR INKEY(47)=0 THEN PEN
0:SOUND 1,0,0,7,1,0,15:PLOT 126,16,0:DR
AWR 191,189,0:PLOT 498,16,0:DRAWR -191,1
89,0:FOR Q=1 TO 10:NEXT:PLOT 126,16,1:DR
AWR 191,189,1:PLOT 498,16,1:DRAWR -191,1
89,1:IF X>15 AND X<21 AND Y>11 AND Y<13
THEN GOTO 530
340 IF JOY(0)=8 OR INKEY(4)=0 THEN LET
X=X+1
350 LET X=(X+RND*2)-RND*2:LET Y=(Y+RND*2
)-RND*2
360 PEN 2: LOCATE 3,1:PRINT "SCORE ";SCO
RE
370 IF X<1 THEN LET X=1
380 IF X>31 THEN LET X=31
390 IF Y<2 THEN LET Y=2
400 IF Y>22 THEN LET Y=22
410 IF X1<>X OR Y1<>Y THEN LOCATE X1,Y1:
PRINT " " " ":LOCATE X1,Y1+1:PRINT
" "
420 LOCATE X,Y:PEN 3:PRINT " ";CHR$(255
):PEN 2
430 LOCATE X,Y+1:PEN 3:PRINT CHR$(254)+C
HR$(254)+CHR$(253)+CHR$(252)+CHR$(252):P
EN 2
440 LET X1=X:LET Y1=Y
450 FOR t=1 TO 10:NEXT
460 LET TICKER=INT((TIME/300)-DATUM)
470 LOCATE 30,1:PRINT "TIME ";250-TICKER
480 WEND
490 CLS:PRINT "SORRY !":PRINT"YOU RAN OU
T OF TIME AND HAVE TO RETURN TO BASE TO

```

```

REFUEL."
500 LOCATE 5,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
510 IF INKEY(18)<>0 THEN GOTO 510
520 RUN
530 IF INT(SCORE)=21 THEN LET SCORE=SCOR
E+5 ELSE LET SCORE=SCORE+2:LOCATE 3,1:PE
N 2:PRINT "SCORE ";SCORE
540 IF SCORE>50 THEN GOTO 560
550 LOCATE X,Y:PRINT "           ":LOCA
TE X,Y+1:PRINT "           ":GOTO
190
560 LOCATE X,Y:PRINT " ";CHR$(238)
570 LOCATE X,Y+1:PRINT CHR$(238);CHR$(23
8);CHR$(238);CHR$(238);CHR$(238)
580 ENV 1,1,0,50,14,-1,10:SOUND 1,0,0,15
,1,0,31:SOUND 2,0,0,15,1,0,15:SOUND 4,0,
0,15,1,0,1
590 FOR T=1 TO 100:NEXT
600 CLS:PRINT "WELL DONE":PRINT"YOU HAVE
DESTROYED THE      ENEMY."
610 LOCATE 5,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
620 IF INKEY(18)<>0 THEN GOTO 620
630 RUN

```


ALIEN BLASTER

Alien beings from another planet are invading Earth with the intention of colonizing it. It only requires five alien vessels to land to establish a colony - so keep blasting. There are ten levels of difficulty in this game, with the alien vessels descending more quickly at higher levels.

Your task is to blast the aliens with your laser gun, before they touch down. The laser base is moved left and right with the Q and P keys, pressing the space bar to fire. You can also use the joystick.

Detailed description

40-70 Install the user-defined graphics characters.

80-120 Set up the ink colours and border.

130 Set the score and high score to zero.

140-200 Clear the screen. Print the instructions for playing the game.

210-260 Clear the screen and invite the player to select the level of difficulty. Check to ensure that the value entered is within the allowed range.

280 Set to five the number of ships required to land in order to win the game. (You could alter this value to prolong or shorten the game.)

290-310 Set the initial values of the variables.

330-350 Clear the screen and print the score and high score.

370-390 Select at random the starting position of the alien ship and the type of ship to be printed.

400-420 Set the counter. Erase the alien ship and randomly move its position left or right.

430-460 Check that the alien has not gone off the screen. Print the alien ship.

470-490 Detect which key has been pressed.

500-520 Check that the laser base will not go off the screen. Print the laser base.

530 Generate a pause, the length of which is determined by the level of difficulty. This controls the speed of the game.
540 If the counter set in line 400 does not exceed 23 then loop back and continue the alien's descent.
550-590 If the alien has landed then increment the variable that records this. Print the total number of aliens that have landed and loop back to line 360.
600-720 Check the score against the high score, and if it is higher, print a congratulations message and play a tune. Offer another game to the player.
730-850 Plot and draw the laser beam and generate the sound of firing. A check is then made to determine if a hit has been scored. If so, print an explosion, update and print the score.

```
10 REM ALIEN BLASTER
20 REM CONVERSION BY TIM BANKS
30 REM ORIGINAL BY ALAN GREEN
40 SYMBOL AFTER 240
50 SYMBOL 255,&7E,&DB,&FF,&7E,&42,&81,&4
2,&24
60 SYMBOL 254,&7E,&7E,&5A,&FF,&FF,&24,&4
2,&81
70 SYMBOL 253,&18,&18,&18,&18,&18,&FF,&8
1,&81
80 INK 0,0
90 INK 1,26
100 INK 2,6
110 INK 3,24
120 BORDER 0
130 LET HIGH=0:LET SCORE=0
140 CLS:LOCATE 15,1:PEN 1:PRINT "ALIEN B
LASTER"
150 LOCATE 1,3:PRINT "THE ALIENS ARE INV
ADING YOUR SYSTEM,      YOUR ONLY HOPE IS
TO BLAST THEM BEFORE THEY LAND AND COLO
NIZE THE EARTH!":PRINT
160 PRINT "WARNING: ONLY FIVE ALIENS HAV
E TO LAND TO TAKE OVER THE PLANET!!!"
170 PRINT:PRINT:PRINT "CONTROLS:":PRINT:
PRINT "KEYBOARD : Q = MOVE BASE LEFT
                  P = MOVE BASE RIGHT":P
```

```

RINT:PRINT "                <SPACE> TO FIRE"
180 PRINT:PRINT"OR USE THE JOYSTICK."
190 LOCATE 5,25:PRINT "PRESS <ENTER> TO
START"
200 IF INKEY (18) <> 0 THEN GOTO 200
210 CLS
220 WHILE LEVEL <1 OR LEVEL >10
230 FOR TY=1 TO 50:BUFFER$=INKEY$:NEXT
240 LOCATE 3,10:PEN 1: INPUT "ENTER LEVE
L OF DIFFICULTY (1-10) ";LEVEL
250 LET LEVEL =11-LEVEL
260 WEND
280 LET SHIPS=5
290 LET A=2
300 LET X=20
310 LET FG=1
330 CLS
340 LOCATE 1,25:PEN 1:PRINT "ALIENS "
350 PEN 1:LOCATE 1,1:PRINT "SCORE ";:PEN
2:PRINT L:PEN 1:LOCATE 20,1:PRINT "HIGH
SCORE ";:PEN 2 :PRINT HIGH
360 WHILE SHIPS>0
370 LET P=INT(RND*40)+1
380 IF P<X-10 OR P>X+10 THEN GOTO 370
390 IF INT(RND*2)+1=2 THEN LET A$=CHR$(2
55) ELSE LET A$=CHR$(254)
400 LET A=A+1
410 LOCATE FG,A-1:PRINT " "
420 IF INT(RND*2)+1=2 THEN LET P=P+1:ELS
E LET P=P-1
430 IF P>39 THEN LET P=39
440 IF P<1 THEN LET P=1
450 LET FG=P
460 LOCATE P,A:PEN 3:PRINT A$:PEN 2
470 IF JOY(0)=16 OR INKEY(47)=0 THEN GOT
O 730
480 IF JOY(0)=4 OR INKEY(67)=0 THEN LET
X=X-1
490 IF JOY(0)=8 OR INKEY(27)=0 THEN LET
X=X+1
500 IF X<1 THEN LET X=1
510 IF X>40 THEN LET X=40
520 LOCATE X,23:PRINT " ";CHR$(253);" "
530 FOR T=1 TO LEVEL*10: NEXT T
540 IF A<23 THEN GOTO 400

```

```

550 LOCATE P,23:PRINT " ":LET A=2
560 LET SHIPS=SHIPS-1
570 LET L=L+1
580 LOCATE 1,25:PEN 1:PRINT "ALIENS ";:P
EN 2:PRINT L
590 WEND
600 IF SCORE > HIGH THEN LET HIGH=SCORE
610 IF SCORE < HIGH THEN GOTO 660
620 CLS
630 ENT 1,30,-10,5:SOUND 3,300,0,15,15,1
640 PRINT "CONGRATULATIONS!"
650 PRINT :PRINT"YOU HAVE A NEW HIGH SCO
RE OF ":PRINT :PRINT HIGH
660 LOCATE 12,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
670 IF INKEY(18)<>0 THEN GOTO 670
680 LET SHIPS=5
690 LET L=0
700 LET LEVEL=0
710 LET score=0
720 GOTO 210
730 PLOT X*16+5,49,1:DRAWR 0,300,1
740 PLOT X*16+5,49,0:DRAWR 0,300,0
750 SOUND 6,20,5,7
760 IF X+1=P THEN GOTO 780
770 GOTO 480
780 LOCATE P,A:INK 3,7:PEN 3:PRINT "*"
790 SOUND 1,0,25,7,0,0,15
800 FOR G=1 TO 10:NEXT
810 INK 3,24
820 PEN 2
830 LOCATE P,A:PRINT " "
840 LET SCORE = SCORE+5:LOCATE 8,1:PRINT
SCORE
850 LET A=2:GOTO 590

```

MINESWEEPER

You are the captain on a minesweeper and your mission is to chart enemy coastal minefields. The mines are not visible but your radar beeps if you move to a mined square and the computer displays the number of mines planted.

Your ship is located on the left-hand side of the screen and your objective is to get to the right, without being blown up. You score one point for each move and an extra fifty points for a successful crossing of the minefield.

If you approach within one move of a mine you must guess where it is and avoid it - one wrong move and it's all hands to the life-boats. To move your ship use the following keys:

QUp
ADown
PRight
OLeft

Good luck and keep sweeping!

Detailed description

30-40 Set the ink, pen, paper and border colours and clear the screen.

50-70 Install the user-defined graphics characters.

80 Set the variables to their start values.

90-190 Print the instructions for playing the game.

200 Disable the interrupt feature.

210 Re-set the pen and paper colours and clear the screen.

220-240 Dimension the replay and minefield arrays and set the counter to zero.

250-310 Locate the mines at random on the screen in the same colour as the paper so that they are not visible. The position of a mine is indicated by the array **SCREEN**.

320 Enable the interrupt feature.

330-350 Clear the screen and set the initial screen co-ordinates of the ship. Print the score, level and number of

lives at the top of the screen.

360-370 Erase the ship from the screen and print it in the new position.

380-390 Increment the move counter and place the screen co-ordinates of the ship into the replay array.

400 If the ship has hit a mine go to line 610 for the explosion.

410-470 Detect how many mines are adjacent to the ship and print this number at the bottom of the screen. Reset MINE to zero. 'Beep' to warn the player of the presence of mines.

480-550 If a key has been pressed then adjust the screen co-ordinates of the ship and up-date the score. The score is only incremented for forward movements; vertical movements are not scored; score is reduced for backward moves. If the ship has not completed the journey across the screen the program returns to line 350 for the next move.

560-600 If the ship has navigated the sea successfully then increment the score by 50 points. The level of play is then increased and printed alongside the score. The game is continued on any keypress.

610-630 Locate the position where the ship has hit a mine, print the explosion character and trigger the sound of the explosion. Operate a short pause.

640-710 Re-set the ink colour. Print the location of all the mines in the minefield.

720-740 Clear the buffer. Invite the player to watch a replay of the ship's journey. A choice of speeds is offered for the replay.

750-840 Using the replay array, the course of the ship is printed from the start until it hit the mine which sank it.

850-880 Ask the player to press any key to continue the game and reduce the number of lives. The screen is cleared and the arrays erased and the program loops back to the start.

890-950 If the player has lost all three lives then print a message, print the score and level. Invite the player to play again.

```

10 REM MINESWEEPER
20 REM T.BANKS
30 INK 0,6:INK 1,1:INK 2,11:INK 3,15
40 PEN 2:PAPER 1:CLS
50 SYMBOL AFTER 240
60 SYMBOL 255,0,0,0,8,200,127,254,0
70 SYMBOL 254,0,130,84,56,124,56,84,130
80 SC=0:LIV=3:LEV=1
90 CLS:LOCATE 17,1:PRINT "MINESWEEPER":P
PRINT:PRINT"IN THIS GAME YOU PLAY THE CA
PTAIN OF A MINESWEEPER.":PRINT:PRINT"YOU
HAVE TO GUIDE YOUR SHIP THROUGH A STR
ETCH OF SEA COVERED IN MINES."
100 PRINT:PRINT"YOU ARE EQUIPPED WITH RA
DAR WHICH TELLS YOU IF THERE ARE ANY MIN
ES ADJACENT TO YOUR SHIP."
110 PRINT:PRINT"IF YOU SUCCESSFULLY CROS
S THE SEA YOU ARE GIVEN 50 BONUS POINT
S AND THEN GIVEN A HARDER STRETCH OF WATE
R TO CROSS."
120 PRINT:PRINT"IF YOU RUN INTO A MINE Y
OU ARE GIVEN THE CHOICE OF SEEING WHERE Y
OU WENT WRONG BY AN ACTION REPLAY OF T
HE COURSE YOU TOOK."
130 LOCATE 5,25:PRINT "PRESS ANY KEY TO
CONTINUE"
140 IF INKEY$="" THEN GOTO 140
150 CLS:PRINT"CONTROLS :":PRINT:PRINT"KE
YBOARD Q-UP
      A-DOWN
      O-LEFT
      P-RIGHT"
160 PRINT:PRINT
170 PRINT "JOYSTICK CONTROL IS AVAILABLE
"
180 LOCATE 5,25:PRINT "PRESS ANY KEY TO
START"
190 IF INKEY$="" THEN GOTO 190
200 DI
210 PEN 2:PAPER 1:CLS
220 DIM REPLAY(2000,2)
230 CLS:DIM SCREEN(40,25)
240 LET COUNT=0
250 FOR A=1 TO (LEV*50)

```

```

260 LET N=INT(RND*40)+1:LET M=INT(RND*22
)+2
270 IF N=1 AND M=13 THEN GOTO 260
280 LET SCREEN(N,M)=1
290 PEN 1
300 LOCATE N,M:PRINT CHR$(254)
310 NEXT
320 EI
330 CLS
340 LET X=1:Y=13:PX=1:PY=13
350 LOCATE 1,1:PEN 3:PRINT "LEVEL ";LEV:
LOCATE 15,1:PRINT "SCORE ";SC:LOCATE 30,
1:PRINT "LIVES ";LIV
360 LOCATE PX,PY:PRINT " "
370 PEN 3:LOCATE X,Y:PRINT CHR$(255)
380 LET PX=X:PY=Y
390 LET COUNT=COUNT+1:LET REPLAY(COUNT,1
)=X:LET REPLAY(COUNT,2)=Y
400 IF SCREEN(X,Y)=1 THEN GOTO 610
410 IF SCREEN(X+1,Y)=1 THEN LET MINE=MINE+1
420 IF SCREEN(X,Y+1)=1 THEN LET MINE=MINE+1
430 IF SCREEN(X,Y-1)=1 THEN LET MINE=MINE+1
440 IF SCREEN(X-1,Y)=1 THEN LET MINE=MINE+1
450 IF MINE>0 THEN PRINT CHR$(7)
460 LOCATE 1,25:PRINT "ADJACENT MINES ";
MINE
470 LET MINE=0
480 LET A$=INKEY$
490 IF A$="" THEN GOTO 480
500 LET A$=UPPER$(A$)
510 IF A$="Q" OR JOY(0)=1 AND Y>1 THEN L
ET Y=Y-1
520 IF A$="A" OR JOY(0)=2 AND Y<24 THEN
LET Y=Y+1
530 IF A$="O" OR JOY(0)=4 AND X>1 THEN L
ET X=X-1:LET SC=SC-2
540 IF A$="P" OR JOY(0)=8 AND X<40 THEN
LET X=X+1:LET SC=SC+2
550 IF X<>40 THEN GOTO 350
560 LET LEV=LEV+1:LET SC=SC+50:ERASE SCR
EEN:ERASE REPLAY:CLS:PRINT "CONGRATULATI

```



```


ONS YOU SUCCESSFULLY FOUND YOUR WAY THROU
UGH THE MINES"
570 PRINT:PRINT"LEVEL ";LEV:PRINT:PRINT"
SCORE ";SC:PRINT:PRINT"LIVES ";LIV
580 LOCATE 5,25:PRINT "PRESS ANY KEY TO
CONTINUE"
590 IF INKEY$="" THEN GOTO 590
600 GOTO 210
610 LOCATE X,Y:INK 0,3,24:PEN 0:PRINT CH
R$(238)
620 ENV 1,1,0,50,14,-1,10:SOUND 1,0,0,15
,1,0,31:SOUND 2,0,0,15,1,0,15:SOUND 4,0,
0,15,1,0,1
630 FOR Y=1 TO 1000:NEXT
640 INK 0,6
650 FOR R=1 TO 25
660 FOR E=1 TO 40
670 PEN 0
680 IF SCREEN(E,R)=0 THEN GOTO 700
690 LOCATE E,R:PRINT CHR$(254)
700 NEXT
710 NEXT
720 FOR GH=1 TO 100:BUFFER$=INKEY$:NEXT:
LOCATE 1,25:PRINT "
":LOCATE 1,25:INPUT "AC
TION REPLAY (Y/N)";B$
730 IF UPPER$(B$)<>"Y" THEN GOTO 870
740 LOCATE 1,25:PRINT "
":LOCATE 1,25:INPUT
"SPEED 1-5";SP:IF SP<1 OR SP>5 THEN GOTO
740
750 PAPER 2
760 LET F=1
770 WHILE F<>0
780 IF REPLAY(F,1)=0 THEN LET F=0:GOTO 8
30
790 IF F=1 THEN GOTO 820
800 LOCATE REPLAY(F-1,1),REPLAY(F-1,2):P
RINT " "
810 LOCATE REPLAY(F,1),REPLAY(F,2):PRINT
CHR$(255)
820 LET F=F+1
830 FOR V=1 TO SP*100:NEXT
840 WEND

```

```

850 LOCATE 1,25:PRINT "
      ":LOCATE 1,25:PRINT
  "PRESS ANY KEY TO CONTINUE"
860 LET F$=INKEY$: IF F$="" THEN GOTO 860
870 LET LIV=LIV-1
880 IF LIV<>0 THEN ERASE SCREEN:ERASE RE
PLAY:GOTO 200
890 CLS
900 PAPER 1:CLS:PEN 3
910 PRINT"HARD LUCK: YOU DIED."
920 PRINT:PRINT"YOU OBTAINED A SCORE OF
";SC
930 PRINT:PRINT"AND REACHED LEVEL ";LEV
940 LOCATE 5,25:PRINT "PRESS ANY KEY TO
PLAY AGAIN"
950 LET K$=INKEY$: IF K$="" THEN GOTO 950
960 RUN

```



You are in a graveyard! The church clock has just struck midnight! There are zombies everywhere! (Fortunately you can choose how many - between one and fifty. But the more there are, the higher your score can be). You must evade them for as long as possible because if you are caught you are immediately 'zombified' - not a pleasant prospect!

The zombies are selected at random to move one at a time and are programmed to move straight towards you! But if they fall into a grave they are destroyed. If *you* fall into a grave you will be 'zombified' too. Your aim is to lure as many zombies as possible to fall into graves. The more you destroy the more you score. The controls are from joystick or the following keys:

QUp
ADown
OLeft
PRight

Detailed description

30-60 Install the user-defined graphics characters.
70 Set the mode and clear the screen.
80 Set the variables to zero.
90-100 Set the ink, paper, pen and border colours.
110-160 Print the instructions on the screen.

170-180 Wait for ENTER to be pressed then clear the screen.

190 Clear the buffer.

200-240 Set the string A\$ to represent the graves then print the graves on the screen.

250 Print the player at the top left-hand corner of the screen.

260 Invite the player to enter the number of zombies required.

270-280 Check that the number of zombies entered is within the permitted range and dimension the array Z to this value.

290-350 Select at random a position on the screen at which to place a zombie. A check is made to ensure the position is not occupied.

360 Clear the text from the bottom of the screen.

370-400 Check for a keypress or joystick movement and then update the screen co-ordinates of the player.

410 If the player has not moved do not reprint the figure on the screen.

420-430 Test to see if the player has fallen into a grave by checking the ink colour at the new location. Go to the end of game routine if the player has fallen in.

440-480 Delete the player's character and reprint at the new location, with the appropriate sound. Go to the sub-routine for moving the zombies.

490-500 Print the current score and loop back to the start of the routine.

510-580 Select a zombie at random and check it has not been killed. If the zombie is still active then move it towards the player.

590 Check if the zombie has fallen into a grave. Check to see if all the zombies have been destroyed.

600-630 Check to determine if the zombie is to move to the position occupied by the player. If so go to the end of game routine. Otherwise print the zombie in the new position with the appropriate sound.

640-710 Clear the buffer. Print a congratulations message and play congratulations sound. Print the score

and high score. The player is then invited to have another game.

720-790 Clear the buffer. Print a losing message with the appropriate tune. Print the score and high score. Invite the player to play again.

800-810 Short pause.

820-830 Sub-routine to clear the buffer.

840 Calculation of high score.

```
10 REM ZOMBIES
20 REM T.BANKS
30 SYMBOL AFTER 240
40 SYMBOL 255,28,8,62,93,28,28,20,20
50 SYMBOL 254,56,16,56,56,56,56,40,40
60 SYMBOL 253,255,243,243,0,0,243,243,25
5
70 MODE 1:CLS
80 COUNT=0:ZOM=0:SCORE=0
90 INK 0,0:INK 1,26:INK 2,7:INK 3,24
100 PAPER 0:CLS:BORDER 0:PEN 1
110 LOCATE 18,1:PRINT "ZOMBIES":PRINT:PR
INT"YOU ARE IN A GRAVEYARD AND THE CHURC
H   CLOCK HAS JUST STRUCK MIDNIGHT !"
120 PRINT:PRINT"YOU MUST AVOID THE ZOMBI
ES FOR AS LONG   AS POSSIBLE BY LURING TH
EM INTO THE      GRAVES."
130 PRINT:PRINT"IF YOU ARE CAUGHT, OR YO
U RUN INTO A     GRAVE, YOU WILL BE ZOMBI
FIED."
140 PRINT:PRINT"CONTROLS :":PRINT:PRINT"
KEYBOARD : Q-UP
          A-DOWN
          O-LEFT
          P-RIGHT"
150 PRINT:PRINT"JOYSTICK CONTROL IS ALSO
AVAILABLE"
160 LOCATE 5,25:PRINT "PRESS <ENTER> TO
CONTINUE"
170 IF INKEY(18)<>0 THEN GOTO 170
180 CLS
190 FOR Q=1 TO 10:LET BU$=INKEY$:NEXT
200 LET A$=CHR$(143)+CHR$(253)+CHR$(143)
210 LET A$="   "+A$+"   "+A$+"   "+A$+"   "
```

```

    "+A$+"    "+A$+"    "+A$
220 FOR A=2 TO 24 STEP 4
230 LOCATE 1,A:PRINT A$
240 NEXT
250 LET X=1:LET Y=1:PX=1:PY=1:LOCATE X,Y
:PEN 3:PRINT CHR$(254)
260 LOCATE 1,25:PRINT "
                                ":LOCATE 1,25:INPUT
"HOW MANY ZOMBIES 1-50";ZOM
270 IF ZOM<1 OR ZOM>50 THEN GOTO 260
280 DIM Z(ZOM,2)
290 FOR B=1 TO ZOM
300 LET D=INT(RND*22)+1:LET E=INT(RND*39
)+1
310 IF E=1 THEN 300
320 IF TEST(E*16-8,400-(D*16))<>0 THEN G
OTO 300
330 LET Z(B,1)=E:LET Z(B,2)=D
340 LOCATE E,D:PEN 2:PRINT CHR$(255)
350 NEXT
360 LOCATE 1,25:PRINT "
                                "
370 IF (JOY(0)=1 OR INKEY(67)=0) AND Y>1
    THEN LET Y=Y-1
380 IF (JOY(0)=2 OR INKEY(69)=0) AND Y<2
4    THEN LET Y=Y+1
390 IF (JOY(0)=4 OR INKEY(34)=0) AND X>1
    THEN LET X=X-1
400 IF (JOY(0)=8 OR INKEY(27)=0) AND X<4
0    THEN LET X=X+1
410 IF X=PX AND Y=PY THEN GOTO 480
420 LET F=TEST(X*16-8,408-(Y*16))
430 IF F<>0 THEN GOTO 720
440 LOCATE PX,PY:PRINT " "
450 LOCATE X,Y:PEN 3:PRINT CHR$(254)
460 SOUND 4,200,3,7
470 LET PX=X:PY=Y
480 GOSUB 510
490 PEN 3:LET SCORE=SCORE+1:LOCATE 1,25:
PRINT "SCORE ";SCORE
500 GOTO 370
510 LET W=INT(RND*ZOM)+1
520 IF Z(W,1)=0 THEN GOTO 510
530 LET A1=Z(W,1):LET A2=Z(W,2)
540 IF X>Z(W,1) THEN LET Z(W,1)=Z(W,1)+1

```

```

550 IF X<Z(W,1) THEN LET Z(W,1)=Z(W,1)-1
560 IF Y>Z(W,2) THEN LET Z(W,2)=Z(W,2)+1
570 IF Y<Z(W,2) THEN LET Z(W,2)=Z(W,2)-1
580 LOCATE A1,A2:PRINT " "
590 IF TEST(Z(W,1)*16-8,408-(Z(W,2)*16))
=1 THEN LET Z(W,1)=0:Z(W,2)=0:LET COUNT=
COUNT+1:IF COUNT=ZOM THEN GOTO 640:ELSE
RETURN
600 IF TEST(Z(W,1)*16-8,408-(Z(W,2)*16))
=3 THEN GOTO 720
610 LOCATE Z(W,1),Z(W,2):PEN 2:PRINT CHR
$(255)
620 SOUND 2,1000,5,7
630 RETURN
640 GOSUB 820
650 CLS:PRINT "CONGRATULATIONS!":PRINT:P
RINT"YOU HAVE DESTROYED ALL THE ZOMBIES"
660 ENT 1,10,-20,5:SOUND 1,300,100,7,0,1
670 PRINT:PRINT"YOU OBTAINED A SCORE OF
";SCORE
680 GOSUB 840
690 PRINT:PRINT "YOUR HIGHEST SCORE IS "
;HSCORE
700 LOCATE 5,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
710 GOTO 790
720 GOSUB 820
730 CLS:PRINT "AAHHHHHH!":PRINT:PRINT"YO
U HAVE BEEN ZOMBIFIED"
740 ENV 1,14,-1,10:SOUND 1,30,150,15,1
750 PRINT:PRINT"YOU OBTAINED A SCORE OF
";SCORE
760 GOSUB 840
770 PRINT:PRINT "YOUR HIGHEST SCORE IS "
;HSCORE
780 LOCATE 5,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
790 IF INKEY(18)<>0 THEN GOTO 790
800 FOR Z=1 TO 255: LET Q$=INKEY$:NEXT
810 ERASE Z:GOTO 30
820 FOR HJ=1 TO 255:LET GY$=INKEY$:NEXT
830 RETURN
840 IF SCORE>HSCORE THEN LET HSCORE=SCOR
E:RETURN
850 RETURN

```

ROCKET ATTACK

In this futuristic game you are the commander of an anti-missile laser, charged with the task of defending a nuclear power station from attack. Varying numbers of missiles descend towards the targets and you have to position your laser sights on each, then ... fire!

The power station will absorb only a certain amount of damage before it 'goes critical'. At this point the power station will change colour on your screen indicating that it is approaching the point of total destruction.

There are five speed levels and five levels of difficulty to choose from in this game, which is controlled by either the keyboard or your joystick. The keyboard controls are:

QLeft
WRight
OUp
KDown
SPACEFire

Detailed description

30 Set the ink, pen, paper and border colours and clear the screen.

40-100 Print the instructions for playing the game.

110-150 Dimension the arrays and set up the variables at their initial values. Set up the sound volume envelope.

160-165 Ask the player to enter the speed and level of play. Check for out of range values.

170-220 Set up the length of the sound envelope according to the speed of play selected.

230-290 Install the user-defined graphics characters.

300-310 Set the string **A\$** to represent the power stations and create string **B\$** to represent all the power stations to be printed.

320-340 Clear the screen and reset the pen colour. Print the power stations.

350 Go to the sub-routine to display the score, level, speed and damage.

360-370 Select at random between one and five missiles to be fired.

380 Set the initial position of the laser sights and print them on the screen.

390-430 Check for a keypress or joystick movement, then update the position of the laser sights.

440-490 Delete the laser sight and reprint in new position. Reset the delay flag and update the position of the missiles.

500-510 Start the sound loop when the missiles are at the top of the screen and trigger the sound of the explosion when they reach the bottom.

520-590 Erase the missiles and print them in their new positions.

600-610 If the missiles have landed then go to the damage routine. If they have not landed then repeat the main program loop.

620-680 Print the laser fire with the appropriate sound and operate a small delay before erasing.

690-730 Check to determine if the laser has hit a target. If so, remove the missile from the array and loop back to the main program.

740-800 If the missiles have landed then increase the level of damage by ten points for each missile and if the damage is within forty points of total destruction change the colour of the power stations. If the amount of damage exceeds the value set by the level variable then end the game.

810-870 Print the score, level, speed and damage values.

880-960 Clear the screen. Print the final score and offer the player another game.

```
10 REM ROCKET ATTACK
20 REM T.BANKS
30 INK 0,26:INK 1,0:INK 2,11:INK 3,15:PA
PER 2:CLS:BORDER 11
40 CLS:PEN 1:LOCATE 15,1:PRINT "ROCKET A
TTACK":PRINT:PRINT"YOU ARE THE COMMANDER
OF AN                ANTI-MISSILE LASER."
```

```

50 PRINT:PRINT"YOU MUST DEFEND THE POWER
  STATION FROM ATTACK BY POSITIONING YOU
  R LASER SIGHTS OVER EACH ROCKET AND FIRI
  NG."
60 PRINT:PRINT"THE NUCLEAR POWER STATION
  CAN ONLY TAKE A CERTAIN AMOUNT OF DAMAG
  E AND WILL      GIVE A WARNING OF DESTRUC
  TION BY          CHANGING COLOUR."
70 PRINT:PRINT"CONTROLS :":PRINT:PRINT"K
  EYBOARD : Q-LEFT
              W-RIGHT
              O-UP
              K-DOWN
              <SPACE>-FIRE"
80 PRINT:PRINT"JOYSTICK CONTROL IS ALSO
  AVAILABLE"
90 LOCATE 5,25:PRINT "PRESS ANY KEY TO C
  ONTINUE"
100 IF INKEY$="" THEN GOTO 100
110 COL=0
120 DIM ATT(5)
130 LET SCORE=0
140 LET CO=2
150 ENV 1,1,0,50,14,-1,10
160 CLS:PEN 1:INPUT "SPEED 1-5 ";S:PRINT
  :INPUT"LEVEL 1-5 ";L
165 IF S>5 OR S<1 OR L>5 OR L<1 THEN PRI
  NT CHR$(7):GOTO 160
170 IF S=5 THEN LET Z=5
180 IF S=4 THEN LET Z=4.5
190 IF S=3 THEN LET Z=4
200 IF S=2 THEN LET Z=3.5
210 IF S=1 THEN LET Z=3
220 ENT 1,220,1,Z
230 SYMBOL AFTER 240
240 SYMBOL 240,8,16,48,40,32,32,255,255
250 SYMBOL 241,&44,&28,&10,&10,&10,&FF,&
  FF,&FF
260 SYMBOL 242,0,0,24,24,24,255,255,255
270 SYMBOL 243,12,32,224,224,144,140,128
  ,255
280 SYMBOL 244,4,4,4,252,252,252,255,255
290 SYMBOL 245,124,16,56,56,56,56,56,16
300 LET A$=CHR$(240)+CHR$(210)+CHR$(241)
  +CHR$(242)+CHR$(210)+CHR$(243)+CHR$(244)

```

```

310 LET B$=CHR$(149)+A$+A$+A$+A$+A$+CHR$(149)
320 CLS
330 PEN COL
340 LOCATE 2,25:PRINT B$
350 GOSUB 810:HITC = 0
360 FOR W=1 TO 5:IF RND>0.5 THEN LET ATT(W)=INT(RND*35)+3:LET HITC = HITC + 1
370 NEXT
380 LET X=20:LET Y=12:LET XP=X:YP=Y:LOCATE X,Y:PEN 3:PRINT "+"
390 IF JOY(0)=16 OR INKEY(47)=0 THEN GOT 0 620
400 IF (JOY(0)=1 OR INKEY(34)=0) AND Y>2 THEN LET Y=Y-1
410 IF (JOY(0)=2 OR INKEY(37)=0) AND Y<2 THEN LET Y=Y+1
420 IF (JOY(0)=4 OR INKEY(67)=0) AND X>1 THEN LET X=X-1
430 IF (JOY(0)=8 OR INKEY(59)=0) AND X<3 THEN LET X=X+1
440 LOCATE XP,YP:PRINT " "
450 LOCATE X,Y:PEN 3:PRINT "+"
460 LET XP=X:LET YP=Y
470 LET CFLAG=CFLAG+1:IF CFLAG<5+2 THEN GOTO 390
480 LET CFLAG=0
490 LET CO=CO+1
500 IF CO=3 THEN SOUND 1,20,Z*220,7,0,1
510 IF CO=25 AND HITC <> 0 THEN SOUND 2,0,0,15,1,0,15
520 FOR T=1 TO 5
530 IF ATT(T)=0 THEN GOTO 550
540 LOCATE ATT(T),CO-1:PRINT " "
550 NEXT
560 FOR U=1 TO 5
570 IF ATT(U)=0 THEN GOTO 590
580 LOCATE ATT(U),CO:PEN 1:PRINT CHR$(245)
590 NEXT
600 IF CO=25 THEN GOTO 740
610 GOTO 390
620 PLOT 24,24:DRAWR (X*16)-30,400-(Y*16)-14,1
630 PLOT 600,24:DRAWR X*16-639+24,400-(Y

```

```

*16)-14,1
640 SOUND 4,40,10,7
650 FOR Q=1 TO 100:NEXT
660 PLOT 24,24,2:DRAWR (X*16)-30,400-(Y*
16)-14,2
670 PLOT 600,24,2:DRAWR X*16-639+24,400-
(Y*16)-14,2
680 LOCATE X,Y:PEN 3:PRINT "+"
690 FOR G=1 TO 5
700 IF ATT(G)=0 THEN GOTO 720
710 IF ATT(G)=X AND CO=Y THEN LET ATT(G)
=0:LET SCORE=SCORE+5:HITC=HITC-1:IF HITC
= 0 THEN SOUND 129,0,0,0
720 NEXT
730 GOTO 390
740 FOR P=1 TO 5
750 IF ATT(P)=0 THEN GOTO 790
760 LET DA=DA+10
770 IF DA=L*100 THEN GOTO 880
780 IF DA>L*100-40 THEN COL=3
790 NEXT
800 CO=2:LOCATE X,Y:PRINT " ":GOTO 330
810 PAPER 1:LOCATE 1,1:PRINT "
"
820 LOCATE 1,1:PEN 3:PRINT "SCORE";SCORE

830 LOCATE 11,1:PRINT "LEVEL";L
840 LOCATE 21,1:PRINT "SPEED";S
850 LOCATE 30,1:PRINT "DAMAGE";DA
860 PAPER 2
870 RETURN
880 FOR F=1 TO 255:LET DF$=INKEY$:NEXT
890 CLS:PEN 1
900 PRINT "THE INSTALLATION HAS BEEN DES
TROYED"
910 PRINT:PRINT"YOU OBTAINED A SCORE OF
";SCORE
920 PRINT:PRINT"ON LEVEL ";L
930 PRINT:PRINT"SPEED FACTOR ";S
940 LOCATE 5,25:PRINT "PRESS ANY KEY TO
PLAY AGAIN"
950 IF INKEY$="" THEN GOTO 950
960 RUN

```

ESCAPE

You are a prisoner in a dreaded PCW camp (Person Caught Without a computer). Your sole objective is to **ESCAPE!** In the dead of night you find your way to the camp perimeter; where, to your horror, you find three zones of electrified wire. The zones are represented by a maze, each more difficult than the last. You must escape each maze without touching the wire.

To move you can use either a joystick or the following keys:

QUp
ADown
OLeft
PRight

Detailed description

30-40 Set the ink, pen, paper and border colours.

50-110 Clear the screen and print the instructions for playing the game.

120 Set the variables at their initial values.

130-280 Print the first level maze and then go to the movement and test routines.

290-520 Reset the pen colour, print the second level maze and then go to the movement and test routines.

530-670 Reset the pen colour, print the third level maze and then go to the movement and test routines.

680-770 Clear the screen. If the player has escaped all the mazes then go to the congratulations routine. If not, then print a failure message with the appropriate sound and restart the program on any keypress.

780-850 Detect if a key has been pressed or the joystick has been moved. Delete the prisoner from his current position and adjust the co-ordinates to the new position.

860 Check to see if the man has run into the wire and if so go to the end of game routine.

870-880 If the wire has not been touched, then print the man and generate the sound of his movement. Return to the current maze.

890-920 Print the congratulations message and invite the player to have another game.

```
10 REM ESCAPE
20 REM T.BANKS
30 INK 0,0:INK 1,26,11:INK 2,15:INK 3,8
40 PAPER 0:CLS:PEN 1:BORDER 0
50 CLS:PEN 2:LOCATE 18,1:PRINT "ESCAPE":
PRINT:PRINT"YOU ARE A PRISONER TRYING TO
  ESCAPE      FROM A MAZE WHICH HAS BEEN E
LECTRIFIED."
60 PRINT:PRINT "YOU MUST PROCEED THROUGH
  THE MAZE      WITHOUT TOUCHING THE SID
ES. IF YOU      TOUCH THE SIDES....ZAP!"
70 PRINT:PRINT"CONTROLS : ":PRINT:PRINT"
KEYBOARD : Q-UP
              A-DOWN
              O-LEFT
              P-RIGHT"
80 PRINT:PRINT"JOYSTICK CONTROL IS ALSO
AVAILABLE"
90 LOCATE 5,25:PRINT "PRESS ANY KEY TO C
ONTINUE"
100 IF INKEY$="" THEN GOTO 100
110 CLS:PEN 1
120 LET X=2:Y=2:IX=1:IY=0:CX=2:CY=2
130 FOR B=1 TO 23 STEP 2
140 LOCATE 1,B
150 FOR A=1 TO 40
160 PRINT "#";
170 NEXT
180 NEXT
190 FOR C=1 TO 23
200 LOCATE 1,C:PRINT "#"
210 LOCATE 40,C:PRINT "#"
220 NEXT
230 FOR D=3 TO 23 STEP 2
240 LOCATE INT(RND*35)+2,D:PRINT "  "
250 NEXT
260 GOSUB 790
270 FOR T=1 TO 200:NEXT
```

```

280 IF Y<24 THEN GOTO 260
290 PEN 1
300 CLS
310 FOR A1=1 TO 23 STEP 2
320 LOCATE 1,A1
330 FOR B1=1 TO 40
340 PRINT "#";
350 NEXT
360 NEXT
370 FOR C1=1 TO 23
380 LOCATE 1,C1:PRINT "#":LOCATE 40,C1:P
RINT "#"
390 NEXT
400 FOR D1=1 TO 21
410 LOCATE 20,D1:PRINT "#"
420 NEXT
430 FOR E=3 TO 21
440 LOCATE INT(RND*16)+2,E:PRINT "  ":NE
XT
450 FOR F=23 TO 3 STEP-1:LOCATE 30,F:PRI
NT "#":NEXT
460 FOR G=3 TO 21 STEP 2:LOCATE INT(RND*
8)+21,G:PRINT "  ":NEXT
470 LOCATE 39,23:PRINT "  "
480 FOR J=3 TO 21:LOCATE INT(RND*8)+31,J
:PRINT "  ":NEXT
490 LET X=2:LET Y=2:LET IY=0:LET IX=2:CX
=2:CY=2
500 GOSUB 790
510 FOR Z=1 TO 200:NEXT
520 IF Y<23 THEN GOTO 500
530 CLS:PEN 1
540 FOR Q1=1 TO 23 STEP 2
550 LOCATE 1,Q1
560 FOR W1=1 TO 40
570 PRINT "#";
580 NEXT :NEXT
590 FOR Q2=1 TO 23:LOCATE 1,Q2:PRINT "#"
:LOCATE 40,Q2:PRINT "#":NEXT
600 FOR W2=1 TO 21:LOCATE 10,W2:PRINT "#
":LOCATE 30,W2:PRINT"#":NEXT
610 FOR Q3=23 TO 3 STEP -1:LOCATE 20,Q3:
PRINT "#":NEXT
620 FOR K1=3 TO 21:LOCATE INT(RND*6)+2,K
1:PRINT "  ":LOCATE INT(RND*6)+11,K1:PRI

```

```

NT " ":LOCATE INT(RND*6)+21,K1:PRINT "
":LOCATE INT(RND*6)+31,K1:PRINT " ":NE
XT
630 LOCATE 39,1:PRINT " "
640 LET X=2:Y=2:IY=0:IX=1:CY=2:CX=2
650 GOSUB 790
660 FOR L=1 TO 200:NEXT
670 IF X>1 THEN GOTO 650
680 CLS
690 IF X=1 AND (Y=39 OR Y=40) THEN GOTO
890
700 FOR V=1 TO 255:LET FG$=INKEY$:NEXT
710 CLS
720 PEN 1:LOCATE 1,1:PRINT"ZZZZZZZZZZZZZ
ZZZZZAAAAAAPP":ENT 1,10,5,5:SOUND 2,30,
50,7,0,1
730 PEN 2
740 PRINT:PRINT"YOU RAN INTO THE FENCE A
ND HAVE BEEN ELECTROCUTED"
750 LOCATE 5,25:PRINT "PRESS ANY KEY TO
PLAY AGAIN"
760 IF INKEY$="" THEN GOTO 760
770 RUN
780 LOCATE 2,2
790 IF (JOY(0)=1 OR INKEY(67)=0) THEN LE
T IX=0:IY=-1
800 IF (JOY(0)=2 OR INKEY(69)=0) THEN LE
T IX=0:IY=1
810 IF (JOY(0)=4 OR INKEY(34)=0) THEN LE
T IX=-1:IY=0
820 IF (JOY(0)=8 OR INKEY(27)=0) THEN LE
T IX=1:IY=0
830 LOCATE CX,CY:PRINT " "
840 LET X=X+IX:LET Y=Y+IY
850 LET CX=X:LET CY=Y
860 IF TEST(X*16-3,410-(Y*16))=1 THEN GO
TO 700
870 LOCATE X,Y:PEN 2:PRINT CHR$(248):PRI
NT CHR$(7)
880 RETURN
890 CLS:PEN 2:PRINT "WELL DONE YOU HAVE
MADE YOUR WAY THROUGHALL THE MAZES "
900 LOCATE 5,25:PRINT "PRESS ANY KEY TO
PLAY AGAIN"
910 IF INKEY$="" THEN GOTO 910
920 RUN

```




This is a computerized version of the standard game of Patience and it demonstrates the Amstrad's excellent graphics capabilities.

The object of the game is to get all the cards into the box on the right-hand side of the screen - called the Foundation. You have to build them up in suits, starting with the Aces, going through the number cards from two to ten and ending with the Kings. You can put a card into the Foundation only if the card which immediately precedes it in the suit is there already.

The cards are dealt into seven columns, with one card in the first, two in the second, three in the third and so on. Only the last card in each column is dealt face up. When a card is moved, the next card in the column is turned over and revealed.

When moving cards from column to column, you can only put black on red and red on black, and in descending numerical order. To move from column to column simply type in the numbers of the columns that you want to move to and from. All visible cards are moved from one column to the other.

If you get stuck and cannot move between columns, don't despair! The remainder of the cards are stored in the Pack and you can get access to them by pressing D for deal. They are dealt them in threes - only every third card is turned over - but you can deal as many times as you like. However, you

will see the same cards each time unless you can make use of one or more, either in the Foundation or columns.

Press P to move a card from the Pack, together with the number of the column it is to go to. You can also move cards from both Pack and columns to the Foundation by pressing F when asked for the destination.

The program automatically carries out a move so you don't need to press ENTER and it rejects any invalid moves too. Here is a summary of the command keys to press:

P Move card from Pack
F Move card to Foundation
1-7 Choose columns by their number. All visible cards are moved from one column to the destination column
D Deal cards in the Pack
X Quit - for when you get stuck and can't make a move! or win!

Detailed description

30-70 Install the user-defined graphics characters.

80-180 Set the variables to their initial values and dimension the arrays.

190-330 Fill the various card handling arrays (pack, display and hidden cards) with values.

340-350 Set the ink, paper, pen and border colours and clear the screen.

360-510 Print the instructions on the screen.

520 Go to the sub-routine to set up the screen.

530-540 Draw the Foundation area and then go to the sub-routine to print the columns of cards.

550-560 Ask the player to make a move.

570-710 Check the move to determine if it is valid. Process the move to take the cards to their selected destinations.

720-740 Check that a red card is being moved on to a black card or vice-versa.

750-830 Convert the card to a value number and check

that it is being moved on to a card which has a value of one unit higher.

840-900 Routine for moving from the pack to the Foundation through sub-routines at lines 1810 and 1480.

910-1010 Routine for moving from the pack to a column through sub-routines at lines 1680, 1560 and 1480.

1020-1220 Routine for moving cards from a column to the Foundation.

1230-1470 Routine for moving cards from one column to another.

1480-1550 Routine for handling the cards in the deck.

1560-1760 Print the screen display.

1770-1800 Select a card at random and ensure that it is not available to be drawn again.

1810-1950 Select and print the appropriate suit for the card selected.

1960-1990 Print the end of game message and invite the player to have another game.

```
10 REM PATIENCE
20 REM T.BANKS
30 SYMBOL AFTER 240
40 SYMBOL 241,&8,&1C,&1C,&3E,&7F,&3E,&8,
&8
50 SYMBOL 242,&8,&1C,&3E,&7F,&3E,&1C,&8,
&0
60 SYMBOL 243,&0,&36,&7F,&7F,&7F,&3E,&1C
,&8
70 SYMBOL 244,&8,&1C,&3E,&7F,&7F,&3E,&8,
&8
80 LET CLUB=1:DIAMOND=27:HEARTS=40:SPADE
S=14
90 LK=1
100 DCC=0
110 LET ST=-2:LET CO=0
120 DIM DISPLAY(7,20)
130 DIM DECK(24)
140 DIM FIN(4,13)
150 DIM HID(21)
160 DIM COL(7)
170 LET INC=2
```

```

180 DIM PACK(52)
190 FOR A=1 TO 7
200 LET COL(A)=A-1
210 NEXT
220 FOR A1=1 TO 7
230 GOSUB 1770
240 LET DISPLAY (A1,1)=CARD
250 NEXT
260 FOR A2=1 TO 24
270 GOSUB 1770
280 LET DECK(A2)=CARD
290 NEXT
300 FOR A3=1 TO 21
310 GOSUB 1770
320 LET HID(A3)=CARD
330 NEXT
340 INK 0,0:INK 1,26:INK 2,6:INK 3,24
350 PAPER 0:CLS:PEN 1:BORDER 0
360 LOCATE 16,1:PRINT "PATIENCE":PRINT:P
RINT"THIS IS A COMPUTERISED VERSION OF T
HE GAME PATIENCE.":PRINT:PRINT"CARDS A
RE DEALT INTO 7 COLUMNS WITH ONE CARD IN
THE 1st COLUMN, TWO IN THE 2nd AND SO
ON. ONLY THE LAST CARD IN EACH"
370 PRINT"COLUMN IS SEEN, THE REMAINDER
BEING DEALT FACE DOWN. WHEN CARDS AR
E REMOVED FROM A COLUMN THE NEXT FACE DO
WN CARD IS TURNED OVER TO BE SEEN."
380 PRINT:PRINT"THE REMAINDER OF THE PAC
K IS HELD BY THEPLAYER. TO OBTAIN ACCESS
TO THESE CARDS THE PLAYER DEALS THEM IN
THREES (I.E. ONLY EVERY THIRD CARD IS
DISPLAYED ON THE SCREEN). HOWEVER YOU
MAY GO THROUGH THE PACK AS OFTEN AS YOU
WISH."
390 LOCATE 5,25:PRINT "PRESS ANY KEY TO
CONTINUE"
400 IF INKEY$="" THEN GOTO 400
410 CLS: LOCATE 16,1:PRINT "PATIENCE":PR
INT:PRINT"THE OBJECT OF THE GAME IS TO A
SSEMBLE ALL THE CARDS IN NUMERICAL AND
SUIT ORDER IN THE FOUNDATION - THE
RECTANGLE ON THE SCREEN - STARTING WITH
THE ACE"
420 PRINT"AND FINISHING WITH THE KING.":

```

```

PRINT:PRINT"CARDS IN THE COLUMNS MUST BE
  ASSEMBLED IN DECREASING ORDER OF VALUE
  AND MUST BE OF ALTERNATE COLOURS."
430 PRINT:PRINT"CARDS MAY BE MOVED AS FO
LLOWS:           From Pack to Foundatio
n (P-F)           From Pack to Column (P
-No.)             Bottom card from Colum
n to Foundation"
440 PRINT"                (No.-F)
                All visible cards from a Col
umn to           another column (No.-No.):PR
INT:PRINT"* NOTE * A KING MAY ONLY BE MO
VED INTO AN EMPTY COLUMN."
450 LOCATE 5,25:PRINT "PRESS ANY KEY TO
CONTINUE"
460 IF INKEY$="" THEN GOTO 460
470 CLS:LOCATE 16,2:PRINT "PATIENCE"
480 PRINT:PRINT"SUMMARY OF CONTROLS":PRI
NT:PRINT"      P = PACK
                F = FOUNDATION
                D = DEAL NEW CARD IN PACK
                X = QUIT THE GAME
                1-7 = COLUMN NUMBER"
490 LOCATE 5,25:PRINT "PRESS ANY KEY TO
CONTINUE"
500 IF INKEY$="" THEN GOTO 500
510 CLS
520 GOSUB 1680
530 PLOT 364,130:DRAW 364,384:PLOT 574,1
30:DRAW 574,384:PLOT 364,130:DRAW 574,13
0:PLOT 364,384:DRAW 574,384
540 GOSUB 1560
550 LOCATE 2,23:PRINT "MOVE FROM   TO   "
560 FOR BU=1 TO 10: LET BUFFER$=INKEY$:N
EXT
565 LET A$=UPPER$(INKEY$)
570 IF A$="X" THEN GOTO 1960
580 IF A$="P" OR A$="D" THEN GOTO 610
590 IF VAL(A$)>0 AND VAL(A$)<8 THEN GOTO
  610
600 GOTO 565
610 IF A$="D" THEN GOTO 1480
620 LOCATE 12,23:PRINT A$
630 LET B$=UPPER$(INKEY$)
640 IF A$=B$ THEN GOTO 550

```

```

650 IF B$="F" THEN GOTO 680
660 IF VAL(B$)>0 AND VAL(B$)<8 THEN GOTO
680
670 GOTO 630
680 LOCATE 17,23:PRINT B$:IF VAL(A$)>0 A
ND VAL(A$)<8 AND VAL(B$)>0 AND VAL(B$)<8
AND A$<>B$ THEN GOTO 1230
690 IF VAL(A$)>0 AND VAL(A$)<8 AND B$="F
" THEN GOTO 1020
700 IF A$="P" AND VAL(B$)>0 AND VAL(B$)<
8 THEN GOTO 910
710 IF A$="P" AND B$="F" THEN GOTO 840
720 IF Y=0 THEN GOTO 750
730 IF X>26 AND Y>26 THEN GOTO 550
740 IF X<27 AND Y<27 THEN GOTO 550
750 IF X>39 THEN LET X=X-39
760 IF X>26 THEN LET X=X-26
770 IF X>13 THEN LET X=X-13
780 IF Y>39 THEN LET Y=Y-39
790 IF Y>26 THEN LET Y=Y-26
800 IF Y>13 THEN LET Y=Y-13
810 IF Y=0 AND X=13 THEN GOTO 830
820 IF Y<>X+1 THEN GOTO 550
830 RETURN
840 IF DECK(DC)=CLUB OR DECK(DC)=DIAMOND
OR DECK(DC)=HEARTS OR DECK(DC)=SPADES T
HEN GOTO 860
850 GOTO 550
860 IF DECK(DC)=CLUB THEN LET CLUB=CLUB+
1:LOCATE 28,2+CLUB:LET PX=CLUB-1:GOSUB 1
810
870 IF DECK(DC)=DIAMOND THEN LET DIAMOND
= DIAMOND +1:LOCATE 31,2+(DIAMOND-26):L
ET PX=DIAMOND-1:GOSUB 1810
880 IF DECK(DC)=HEARTS THEN LET HEARTS=H
EARTS+1:LOCATE 34,2+(HEARTS-39):LET PX=H
EARTS-1:GOSUB 1810
890 IF DECK(DC)=SPADES THEN LET SPADES =
SPADES +1:LOCATE 25,2+(SPADES-13):LET PX
=SPADES-1:GOSUB 1810
900 LET DECK(DC)=0:GOTO 1480
910 LET X=DECK(DC)
920 LET U=DECK(DC)
930 FOR B1=1 TO 13

```

```

940 IF DISPLAY(VAL(B$),B1)=0 THEN GOTO 9
70
950 LET Y=DISPLAY(VAL(B$),B1)
960 NEXT:PRINT"ERROR":END
970 IF B1=1 THEN LET Y=0
980 GOSUB 720
990 LET DISPLAY(VAL(B$),B1)=U
1000 LET DECK(DC)=0
1010 GOSUB 1680:GOSUB 1560:GOTO 1480
1020 REM PRINTING OF CARD IN FOUNDATION
1030 FOR JK=1 TO 13
1040 IF DISPLAY(VAL(A$),JK)=0 THEN GOTO
1080
1050 LET X=DISPLAY(VAL(A$),JK)
1060 NEXT
1070 PRINT"ERROR":END
1080 IF X=SPADES OR X=DIAMOND OR X=HEART
S OR X=CLUB THEN GOTO 1100
1090 GOTO 550
1100 IF X=CLUB THEN LET CLUB=CLUB+1:LOCA
TE 28,2+CLUB:LET PX=CLUB-1:GOSUB 1810
1110 IF X=DIAMOND THEN LET DIAMOND = DIA
MOND +1:LOCATE 31,2+(DIAMOND-26):LET PX=
DIAMOND-1:GOSUB 1810
1120 IF X=HEARTS THEN LET HEARTS=HEARTS+
1:LOCATE 34,2+(HEARTS-39):LET PX=HEARTS-
1:GOSUB 1810
1130 IF X=SPADES THEN LET SPADES =SPADES
+1:LOCATE 25,2+(SPADES-13):PX=SPADES-1:
GOSUB 1810
1140 IF DISPLAY(VAL(A$),JK-1)=0 THEN GOT
O 1210
1150 IF COL(VAL(A$))=0 THEN LET DISPLAY(
VAL(A$),JK-1)=0:GOTO 1210
1160 IF DISPLAY(VAL(A$),JK-2)<>0 THEN LE
T DISPLAY(VAL(A$),JK-1)=0:GOTO 1210
1170 LET DISPLAY(VAL(A$),JK-1)=HID(LK):I
F HID(LK)=0 THEN LET LK=LK+1
1180 IF LK=22 THEN GOTO 1210
1190 IF DISPLAY(VAL(A$),JK-1)=0 THEN GOT
O 1170
1200 LET HID(LK)=0:LET COL(VAL(A$))=COL(
VAL(A$))-1:IF COL(VAL(A$))<0 THEN LET CO
L(VAL(A$))=0

```

```

1210 LET LK=1:GOSUB 1680:GOSUB 1560
1220 GOTO 550
1230 LET X=DISPLAY(VAL(A$),1)
1240 FOR B=1 TO 13
1250 IF DISPLAY(VAL(B$),B)=0 THEN GOTO 1
280
1260 LET Y=DISPLAY(VAL(B$),B)
1270 NEXT
1280 IF B=1 THEN LET Y=0
1290 GOSUB 720
1300 FOR Q=1 TO 13
1310 IF DISPLAY(VAL(B$),Q)=0 AND Q=1 THE
N GOTO 1350
1320 IF DISPLAY(VAL(B$),Q)=0 THEN GOTO 1
350
1330 NEXT
1340 PRINT"ERROR":END
1350 FOR W=Q TO 13
1360 LET COUNT=COUNT+1
1370 LET DISPLAY(VAL(B$),W)=DISPLAY(VAL(
A$),COUNT)
1380 NEXT
1390 LET COUNT=0
1400 FOR R=1 TO 13
1410 LET DISPLAY(VAL(A$),R)=0:NEXT
1420 FOR P=1 TO 21
1430 IF HID(P)=0 THEN NEXT P:LET DISPLAY
(VAL(A$),1)=0:GOTO 1460
1440 IF COL(VAL(A$))=0 THEN GOTO 1470
1450 LET DISPLAY(VAL(A$),1)=HID(P):LET H
ID(P)=0
1460 LET COL(VAL(A$))=COL(VAL(A$))-1:IF
COL(VAL(A$))<0 THEN LET COL(VAL(A$))=0
1470 GOSUB 1680:GOSUB 1560:GOTO 550
1480 REM DECK
1490 LET DC=DC+3
1500 IF DC>24 THEN LET DC=DC-24
1510 FOR Y1=1 TO 24:IF DECK(Y1)=0 THEN N
EXT:LOCATE 1,25:PRINT "
":GOTO 550
1520 LET Y1=0
1530 IF DECK(DC)=0 THEN LET DC=DC+1:GOTO
1500
1540 LOCATE 2,25:LET PX=DECK(DC):GOSUB 1
810

```



```

1550 GOTO 550
1560 FOR A6=1 TO 7
1570 LET CO=0
1580 LET ST=ST+3
1590 FOR A7=COL(A6) TO 19
1600 LOCATE ST+1,A7+3
1610 LET CO=CO+1
1620 IF DISPLAY(A6,CO)=0 THEN PRINT "  ":
GOTO 1640
1630 LET PX=DISPLAY(A6,CO):GOSUB 1810
1640 NEXT
1650 NEXT
1660 LET CO=0:LET ST=-2
1670 RETURN
1680 LOCATE 2,1:PRINT "1  2  3  4  5  6
7"
1690 FOR A4=2 TO 7
1700 LET INC=INC+3
1710 FOR A5=1 TO COL(A4)
1720 PEN 2:LOCATE INC,A5+2:PRINT CHR$(20
7);CHR$(207):PEN 1
1730 NEXT
1740 NEXT
1750 LET INC=2
1760 RETURN
1770 CARD=INT(RND*52)+1
1780 IF PACK(CARD)=1 THEN GOTO 1770
1790 LET PACK(CARD)=1
1800 RETURN
1810 IF PX>39 THEN LET S$=CHR$(243):LET
H=PX-39
1820 IF PX>26 AND PX<40 THEN LET S$=CHR$
(242):LET H=PX-26
1830 IF PX>13 AND PX<27 THEN LET S$=CHR$
(244):LET H=PX-13
1840 IF PX>0 AND PX<14 THEN LET S$=CHR$(
241):LET H=PX
1850 LET H$=STR$(H)
1860 IF H=13 THEN LET H$=" K"
1870 IF H=12 THEN LET H$=" Q"
1880 IF H=11 THEN LET H$=" J"
1890 IF H=10 THEN LET H$=" T"
1900 IF H=1 THEN LET H$=" A"
1910 IF PX>26 THEN PEN 2:ELSE PEN 0
1920 LET H$=RIGHT$(H$,1)

```

```
1930 PAPER 1:PRINT H$;S$
1940 PAPER 0:PEN 1
1950 RETURN
1960 CLS:PRINT "END OF GAME"
1970 LOCATE 5,25:PRINT "PRESS ANY KEY TO
      PLAY AGAIN"
1980 IF INKEY$="" THEN GOTO 1980
1990 RUN
```

ASTEROIDS

As the pilot of a space ship travelling through the universe, you meet a succession of space mines, photon torpedoes, asteroids and other alien space craft. The object of the game is to guide your ship safely through these various hazards.

If a collision is unavoidable you may decide to destroy the obstacle with your laser. The laser has a limited fire power and it should therefore be used sparingly, and only in an absolute emergency. There are four levels of difficulty which follow each other automatically - the number and shape of the obstacles changing at each level.

You can control your spacecraft with either joystick or from the keyboard using these keys on the main keyboard:

1Left
ORight
SPACEFire

Detailed description

30 Set the mode and the ink, border and paper colours for printing the instructions.

40-140 Print the instructions for playing the game.

150-180 Install the user-defined graphics characters.

190 Reset the ink, border and paper colours and clear the screen.

200-240 Set the variables at their initial values.

250 Start of the main loop for playing the game at the first level.

260-270 Scan the keyboard for entries.

280-290 Check to ensure that the space ship does not go off the screen.

300 Check if the space ship has run into an obstacle. If so, go to losing end of game routine.

310 Print the spaceship.

320-340 Print the obstacles at random on the bottom line of the screen.

350 Force hardware scroll. This moves all the obstacles one line up the screen.
360 Check the keyboard and the joystick to see if the laser has been fired. If so go to laser firing routine.
370 Increment the count.
380 End of control loop.
390-410 Create new obstacles and increment the level of difficulty.
420-530 Repeat the main control loop to play the game at the next level of difficulty.
540-550 Check the level and return to the main loop.
560-610 Print a congratulations message and ask the player to play again.
620-670 Print the laser fire and generate the appropriate sound.
680-780 End of game routine if the player has lost. Print a message and play a tune. Invite the player to have another game.

```

10 REM ASTEROIDS
20 REM T.BANKS
30 MODE 1:CLS:INK 0,0:BORDER 0:PAPER 0
40 LOCATE 16,1:PRINT "ASTEROIDS":LOCATE
1,3:PRINT"YOU ARE THE PILOT OF A SPACE S
HIP          TRAVELLING THROUGH THE UNIVERS
E."
50 PRINT:PRINT"YOU MEET A SUCCESSION OF
SPACE MINES,   PHOTON TORPEDOES, ASTEROI
DS AND ALIEN   SPACE CRAFT."
60 PRINT:PRINT"THE OBJECT OF THE GAME IS
SAFELY TO      GUIDE YOUR SHIP THROUGH T
HESE VARIOUS   HAZARDS.":PRINT:PRINT"IF
A COLLISION IS UNAVOIDABLE YOU MAY   DES
TROY THE OBSTACLE WITH YOUR LASER."

65 PRINT:PRINT "YOUR LASER'S ENERGY IS L
IMITED, SO USE  IT SPARINGLY."
70 PRINT:PRINT"THERE ARE 4 LEVELS OF DIF
FICULTY - THE  NUMBER OF OBSTACLES INCRE
ASING AT EACH  STAGE."
80 LOCATE 5,25:PRINT "PRESS <ENTER> TO C
ONTINUE"
  
```

```

90 IF INKEY(18)<>0 THEN GOTO 90
100 CLS
110 PRINT"CONTROLS ":"PRINT:PRINT"KEYBOA
RD - 0 STEER RIGHT
    1 STEER LEFT"
120 PRINT"                <SPACE BAR> TO FIRE
"
130 PRINT:PRINT"OR USE A JOYSTICK.":LOCA
TE 5,25:PRINT "PRESS <ENTER> TO START"
140 IF INKEY(18)<>0 THEN GOTO 140
150 SYMBOL AFTER 240
160 SYMBOL 255,&3C,&7E,&FE,&FF,&FF,&7E,&
1C,&18
170 SYMBOL 254,&18,&18,&3C,&3C,&3C,&3C,&
3C,&3C
180 SYMBOL 253,&3C,&7E,&FF,&FF,&FF,&42,&
42,&0
190 INK 0,0:BORDER 0:PAPER 0:CLS:PEN 1
200 LET BULLETS=50
210 LET LEVEL=1
220 LET X=10
230 LET A$=CHR$(42)
240 LET SCREEN = 1
250 WHILE A<500
260 IF JOY(0)=8 OR INKEY(32)=0 THEN LET
X=X+1
270 IF JOY(0)=4 OR INKEY(64)=0 THEN LET
X=X-1
280 IF X<1 THEN LET X=1
290 IF X>39 THEN LET X=39
300 IF TEST ((X*16)-8,392) <> 0 THEN GOT
O 680
310 LOCATE X,1:PEN 2::PRINT CHR$(228):PE
N 1
320 FOR U=1 TO LEVEL
330 LOCATE INT(RND*40)+1,24:PRINT A$
340 NEXT
350 LOCATE 40,25:PRINT "  "
360 IF JOY(0)=16 OR INKEY(47)=0 THEN GOS
UB 620
370 LET A=A+1:LET SCORE=SCORE+5
380 WEND
390 LET SCREEN=SCREEN +1:IF SCREEN = 2 T
HEN LET A$=CHR$(239):LET A=0:GOTO 250

```

```

400 IF SCREEN = 3 THEN LET A$=CHR$(255):
LET A=0:GOTO 250
410 LET A=0
420 WHILE A<500
430 IF JOY(0)=8 OR INKEY(32)=0 THEN LET
X=X+1
440 IF JOY(0)=4 OR INKEY(64)=0 THEN LET
X=X-1
460 LOCATE X,1:PEN 2:PRINT CHR$(228):PEN
1
470 FOR H=1 TO LEVEL
480 LET F=INT(RND*40)+1:LOCATE F,23:PRIN
T CHR$(254):LOCATE F,24:PRINT CHR$(253)
490 NEXT
500 LOCATE 40,25:PRINT " "
510 IF INKEY(47)=0 OR JOY(0)=16 THEN GOS
UB 620
520 LET A=A+1:LET SCORE=SCORE+5
530 WEND
540 LET LEVEL=LEVEL+1:LET A=0:LET BULLET
S=50
550 IF LEVEL=2 OR LEVEL=3 OR LEVEL=4 THE
N LET A=0 :LET A$="*":GOTO 240
560 CLS
570 LOCATE 12,1: PRINT "CONGRATULATIONS"
580 PRINT:PRINT:PRINT"YOU HAVE CLEARED A
LL FOUR SECTORS"
590 LOCATE 5,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
600 IF INKEY(18)<>0 THEN GOTO 600
610 RUN
620 IF BULLETS <1 THEN RETURN
630 PLOT X*16-8,384,3:DRAWR 0,-16,3:FOR
Q=1 TO 10:NEXT:PLOT X*16-8,384:DRAWR 0,-
16,0:PEN 1
640 SOUND 6,20,5,7
650 SOUND 6,10,5,0
660 LET BULLETS=BULLETS-1
670 RETURN
680 LOCATE X,1:PRINT CHR$(238)
690 ENT 1,30,20,5
700 SOUND 3,300,0,15,15,1
710 FOR V=1 TO 2000
720 NEXT
730 CLS

```

```
740 PRINT "HARD LUCK YOU WERE BLOWN UP"  
750 PRINT: PRINT: PRINT"YOU ACHIEVED A S  
CORE OF ";SCORE  
760 LOCATE 5,25:PRINT "PRESS <ENTER> TO  
PLAY AGAIN"  
770 IF INKEY(18)<>0 THEN GOTO 770  
780 RUN 190
```

BOMBER

This game is an adaptation of the classic arcade game known as 'City Bomber'. You are the pilot of disabled aircraft which is constantly losing height and so must land immediately.

As you are over a city, your only chance is to clear a runway by flattening the buildings with bombs and machine gun fire! You can use a joystick or keyboard control. The joystick button or I on the main keyboard drops a bomb and pushing the stick forward or pressing key O fires the guns.

Detailed description

30-40 Set the pen colour and mode. This program uses the default colours and therefore it is not necessary to set ink colours.

50 Go to the sub-routine to print the instructions for playing the game.

60 Disable the interrupts.

70-100 Install the user-defined graphics characters.

110-120 Reset the pen colour and clear the screen.

130-160 Print the first set of buildings with pen 1.

170-210 Print the second set of buildings with pen 2.

220 Reset the pen colour. Print text on the bottom line of the screen.

230-240 Set the number of bombs available at 250 and the number of bullets at 400. (You may alter these values to make the game more or less difficult.) Enable the interrupts.

250 Check to see if the bomber has collided with a building and if so go to the appropriate end of game routine.

260 Erase the bomber from its current position.

270 Increment the movement of the bomber along the horizontal. If the bomber has reached the end of the line then increment the vertical component to make it lower.

280 Check if the bomber has landed and if so go to the

winning routine.

290 Print the bomber in the new position.

300 Go to the keyboard scan routine.

310 Return to line 250 for the bomber to continue.

320-360 Scan the keyboard for entries and go to the appropriate routine for dropping a bomb or firing the guns.

370-480 Sub-routine for dropping the bombs with the required sound effects. Decrement bomb count.

490-550 Sub-routine for firing the guns. Decrement bullet count.

560-570 Print the losing message and play a tune. Go to the sub-routine offering the player another game.

580-590 Print a congratulations message.

600-620 Invite the player to play again.

630-700 Print the instructions for playing the game.

```
10 REM BOMBER
20 REM T.BANKS
30 PEN 1
40 MODE 1
50 GOSUB 630
60 DI
70 SYMBOL AFTER 240
80 SYMBOL 255,&80,&80,&C6,&FF,&18,&30,&4
0,&0
90 SYMBOL 253,&FF,&FF,&E7,&E7,&E7,&E7,&F
F,&FF
100 SYMBOL 251,&FF,&FF,&FF,&99,&99,&99,&
FF,&FF
110 PEN 1
120 CLS
130 FOR C = 2 TO 38 STEP 2
140 FOR B = INT(RND*10)+10 TO 22
150 LOCATE C,B:PRINT CHR$(253):NEXT B
160 NEXT C
170 PEN 2
180 FOR D = 3 TO 39 STEP 2
190 FOR E = INT(RND*10)+10 TO 22
200 LOCATE D,E:PRINT CHR$(251):NEXT E
210 NEXT D
220 PEN 3:LOCATE 1,24:PRINT "BOMBS ","BU
LLETS "
```

```

230 LET X=1:LET Y=1:LET BOMBS=250:LET BU
LLETS=400
240 LOCATE 7,24:PRINT BOMBS:LOCATE 29,24
:PRINT BULLETS:EI
250 IF TEST((X)*16,400-(Y*16))<>0 THEN G
OTO 560
260 LOCATE X,Y:PRINT " "
270 LET X = X+1:IF X=40 THEN LOCATE X,Y:
PRINT " ": LET Y = Y+1:LET X=1
280 IF Y=23 THEN GOTO 580
290 LOCATE X,Y:CALL &BD19:PRINT CHR$(255
)
300 GOSUB 330
310 GOTO 250
320 IF JOY(0) <> 16 AND JOY(0)<>1 AND IN
KEY(64)<>0 AND INKEY(32)<>0 THEN RETURN
330 IF FLAG=1 THEN LET FLAG=0:RETURN
340 IF JOY(0)= 16 OR INKEY(64)=0 THEN GO
SUB 370
350 IF JOY(0)=1 OR INKEY(32)=0 THEN GOSU
B 490
360 RETURN
370 IF BOMBS<1 THEN RETURN
380 LET Q=Y
390 LET Q=Q+1
400 IF Q=22 THEN GOTO 420
410 IF TEST((X*16)-1,400-((Q)*16))=0 THE
N LOCATE X,Q:PRINT CHR$(252):LOCATE X,Q:
PRINT " ":GOTO 390
420 LOCATE X,Q:PEN 15:PRINT "*"
430 SOUND 1,0,20,7,0,0,30
440 FOR T=1 TO 10:NEXT T
450 LOCATE X,Q:PEN 3:PRINT " "
460 LET FLAG=1
470 LET BOMBS=BOMBS-1:LOCATE 7,24:PRINT
BOMBS
480 RETURN
490 IF BULLETS < 1 THEN RETURN
500 LET W=X
510 SOUND 6,1000,5,7,0,0,0
520 PLOT W*16,408-(Y*16):DRAW 32,0
530 IF TEST((W*16)+1,401-(Y*16))<>0 THEN
LOCATE W+1,Y:PRINT "*":LOCATE W+1,Y:PRI
NT " ":LET BULLETS=BULLETS-2:LOCATE 29,2
4:PRINT BULLETS:RETURN

```

```

540 LET BULLETS=BULLETS-2:LOCATE 29,24:PRINT BULLETS
550 RETURN
560 REM END
570 MODE 0:PEN 15:LOCATE 6,12:PRINT"YOU
CRASHED":ENT 1,30,20,5:SOUND 3,300,0,15,
15,1,0:FOR U=1 TO 4000:NEXT U:MODE 1:GOTO 600
580 MODE 0:CLS:LOCATE 5,13:PEN 15:PRINT
"YOU SURVIVED":FOR E=1 TO 4000:NEXT E
590 MODE 1:CLS
600 PRINT "PRESS <ENTER> TO PLAY AGAIN."
610 IF INKEY(18)<>0 THEN GOTO 610
620 RUN 110
630 CLS:PEN 1:LOCATE 17,1:PRINT "BOMBER"
640 PRINT:PRINT"THIS IS A VERSION OF THE
CLASSIC ARCADE GAME 'CITY BOMBER'."
650 PRINT:PRINT"YOU ARE THE PILOT OF A
DISABLED BOMBER.":PRINT:PRINT"THE ONLY WAY
YOU CAN LAND IS BY CLEARING A RUNWAY WITH
BOMBS AND MACHINE GUN FIRE":PRINT:PRINT
"CONTROLS:":PRINT:PRINT"KEYBOARD: 1-
DROP BOMB":PRINT"                                0-FIRE GUNS"
:
660 PRINT:PRINT "JOYSTICK: ":PRINT:PRINT
T"      FIRE BUTTON TO DROP BOMB":PRINT "
FORWARD TO FIRE GUNS"
670 LOCATE 5,25:PRINT "PRESS <ENTER> TO
START"
680 IF INKEY(18)<>0 THEN GOTO 680
700 RETURN

```



This is simple 'cat and mouse' game. In the centre of the screen is a piece of cheese, and the mouse moves around the screen taking bites out of the cheese as it passes through it. By controlling the cat you can catch the mouse and stop it getting to the cheese.

It is, however, immediately replaced by another mouse, but the more mice you catch the longer the cheese will last. You can use a joystick or these keys:

QUp
ZDown
ILeft
PRight

Detailed description

- 40-120 Print the instructions on the screen.
- 130-150 Set the variables to their initial values.
- 160-190 Set up the ink and border colours.
- 200 Set the time to zero.
- 210 Clear the screen.
- 220-240 Install the user-defined graphics characters.
- 250 Set the pen to yellow to print the cheese.
- 260 Define the cheese as A\$.
- 270-300 Print the cheese on the screen.
- 310 Set the cat's co-ordinates to their initial values.

320 Erase the cat.
 330 Generate the sound of the mouse moving.
 340-370 Scan the keyboard for entries. If a move has
 been made then increment the cat's co-ordinates
 accordingly. Generate the sound of the cat moving.
 380 Check that the cat has not gone off the screen.
 390 Check if the cat has moved on to the cheese, and
 ensure the cheese is not erased.
 400 Check if the cat has caught the mouse.
 410 Print the cat at the current location.
 420 Update and print the time elapsed.
 430 Go to the sub-routine for moving the mouse.
 440 Go to line 320 for next move.
 450 Erase the mouse.
 460-500 Routine for calculating the movement of the
 mouse.
 510 Check if the mouse has got to the cheese.
 520 Check to see if all the cheese has been eaten by the
 mouse.
 530 Check if the mouse has moved on to the cat.
 540-550 Check to ensure that the mouse has not moved
 off the screen.
 560 Print the mouse at the current position.
 570 Loop back to line 430.
 580-610 Generate the sound for the end of the game.
 Print a message and invite the player to have another game.

```

10 REM CHEESE
20 REM CONVERSION BY T.BANKS
30 REM ORIGINAL BY P.BRADSHAW
40 CLS:LOCATE 17,1:PRINT "CHEESE"
50 PRINT:PRINT"THIS IS A SIMPLE 'CAT AND
  MOUSE' GAME!":PRINT:PRINT"THE OBJECT OF
  THE GAME IS TO KEEP THE  MICE FROM THE
  CHEESE FOR AS LONG AS  POSSIBLE."
60 PRINT:PRINT"CATCH THE MICE TO PREVENT
  THEM FROM  EATING THE CHEESE."
70 LOCATE 5,25:PRINT "PRESS <ENTER> TO C
  ONTINUE"
80 IF INKEY(18)<>0 THEN GOTO 80
  
```

```

90 CLS:PRINT "CONTROLS :":PRINT:PRINT"KE
YBOARD- Q UP
      Z DOWN
      P RIGHT"
100 PRINT"          I LEFT":PRINT:PRINT"
JOYSTICK CONTROL IS AVAILABLE."
110 LOCATE 5,25:PRINT "PRESS <ENTER> TO
START"
120 IF INKEY(18)<>0 THEN GOTO 120
130 LET DX=1:LET DY=24
140 COUNT = 0
150 LET Q=1:LET W=-1
160 BORDER 3
170 INK 2,26
180 INK 3,0
190 INK 0,14
200 LET DATUM=INT(TIME/300)
210 CLS
220 SYMBOL AFTER 240
230 SYMBOL 255,&42,&7E,&42,&A5,&81,&99,&
E7,&18
240 SYMBOL 254,&4,&2,&4,&E,&37,&7F,&F,&1
E
250 PEN 1
260 LET A$=CHR$(143)+CHR$(143)+CHR$(143)
+CHR$(143)+CHR$(143)+CHR$(143)+CHR$(143)
270 LOCATE 18,10:PRINT A$
280 LOCATE 18,11:PRINT A$
290 LOCATE 18,12:PRINT A$
300 LOCATE 18,13:PRINT A$
310 LET X=5:LET Y=2
320 LOCATE X,Y:PRINT " "
330 SOUND 1,25,2,5
340 IF JOY(0)=1 OR INKEY(67)=0 THEN IF Y
>1 THEN LET Y=Y-1:SOUND 2,250,2,5
350 IF JOY(0)=2 OR INKEY(71)=0 THEN IF Y
<23 THEN LET Y=Y+1:SOUND 2,250,2,5
360 IF JOY(0)=4 OR INKEY(35)=0 THEN IF X
>1 THEN LET X=X-1:SOUND 2,250,2,5
370 IF JOY(0)=8 OR INKEY(27)=0 THEN IF X
<40 THEN LET X=X+1:SOUND 2,250,2,5
380 IF Y=1 THEN LET Y=2
390 IF TEST(X*16-8,408-(Y*16))=1 THEN GO
TO 340

```

```

400 IF TEST(X*16-3,402-(Y*16))=2 THEN PR
INT CHR$(7):LET DX=INT(RND*40)+1:GOTO 32
0:SOUND 4,0,4,0,7,0,0,1
410 LOCATE X,Y:PEN 3:PRINT CHR$(255):PEN
1
420 LET TICKER =INT((TIME/300)-DATUM):LO
CATE 1,1:PRINT "TIME ";TICKER
430 GOSUB 450
440 GOTO 320
450 LOCATE DX,DY:PRINT " ":LET DX=DX+Q
460 LET DY=DY+W
470 IF DY=3 THEN LET W=1 :IF INT(RND*2)+
1=1 THEN LET DY=4
480 IF DY=25 THEN LET W=-1:IF INT(RND*2)
+1=1 THEN LET DY=24
490 IF DX=1 THEN LET Q=1 :IF INT(RND*2)+
1=1 THEN LET DX=2
500 IF DX=39 THEN LET Q=-1:IF INT(RND*2)
+1=1 THEN LET DX=38
510 IF TEST(DX*16-8,408-(DY*16))=1 THEN
LET COUNT=COUNT+1
520 IF COUNT =28 THEN GOTO 580
530 IF TEST(X*16-3,402-(Y*16))=0 THEN P
RINT CHR$(7):LET DX=INT(RND*40)+1:GOTO 3
20:SOUND 4,0,4,0,7,0,0,1
540 IF DX=0 THEN LET DX=1
550 IF DY=0 THEN LET DY=1
560 LOCATE DX,DY:PEN 2:PRINT CHR$(254)
570 RETURN
580 ENT 1,30,20,5:SOUND 3,300,0,15,15,1:
CLS:PRINT"BAD LUCK ! THE MOUSE HAS WON"
590 PRINT:PRINT"YOU HAD A TIME OF ";TICK
ER
600 PRINT:PRINT"          PRESS <ENTER> TO
PLAY AGAIN."
610 IF INKEY(18)<>0 THEN GOTO 610
620 RUN

```

GRAND PRIX

This is an original game simulating a Grand Prix racing track. You are the driver of the racing car and must steer the car through the track without coming into contact with the chequered wall. There are three levels of difficulty, starting with Formula 3 which is the easiest.

If you are successful in reaching the finishing line, you then go on to the Formula 2 track which is narrower and longer. Success at this level takes you on to the Formula 1 track. Use key 1 on the main keyboard to steer you left, and key 0 to steer right. Joystick control is also available. Drive safely!

Detailed description

30 Set up the ink colours for printing the instructions.

40-130 Clear the screen and print the instructions for playing the game.

140-150 Reset the ink colours.

160-190 Install the user-defined graphics characters.

200 Clear the screen.

210-240 Set the variables at their initial values.

250 Start of the main loop for controlling the game.

260-270 Scan the keyboard for entries.

280-300 Check if the car has run into the wall. If so, then go to the crash routine.

310 Check to see if the car has reached the finishing line.

320 Print the car.

330-370 Move the position of the racing track at random and print it on the bottom line of the screen.

380 Force hardware scroll. This moves the whole of the track one line up on the screen.

390 Delete the finishing line.

400 Increment the count.

410 End of the main control loop.

420-550 Increment the level of difficulty. Proceed to

repeat the main loop at the higher level after printing a suitable message.

560-630 Crash routine. Generate the sound of the crash and print the exploded car. Invite the player to have another game.

```
10 REM GRAND PRIX
20 REM ORIGINAL PROGRAM BY T.BANKS
30 INK 1,24:INK 0,1:INK 2,20:INK 3,6
40 CLS:LOCATE 17,1:PRINT "GRAND PRIX":PR
INT:PRINT "THIS IS AN ORIGINAL GAME SIMU
LATING A  GRAND PRIX RACE TRACK."
50 PRINT:PRINT "YOU ARE THE DRIVER OF TH
E RACING CAR ANDMUST STEER THE CAR THROU
GH THE TRACK  WITHOUT HITTING THE CHEQ
UERED WALL."
60 PRINT:PRINT"THERE ARE 3 LEVELS OF PLA
Y, STARTING  WITH FORMULA 3 WHICH IS T
HE EASIEST."
70 PRINT:PRINT"IF YOU ARE SUCCESSFUL THE
N YOU PROCEED TO FORMULA 2, AND THEN TO
FORMULA 1  WHICH IS THE MOST DIFFICU
LT."
80 LOCATE 5,25:PRINT"PRESS <ENTER> TO CO
NTINUE"
90 IF INKEY(18)<>0 THEN GOTO 90
100 CLS:LOCATE 9,1:PRINT"OPERATING INSTR
UCTIONS":LOCATE 1,4:PRINT"KEYBOARD: 1=ST
EER LEFT 0=ST
EER RIGHT"
110 LOCATE 1,7:PRINT"OR USE A JOYSTICK."
120 LOCATE 5,25:PRINT "PRESS <ENTER> TO
START GAME"
130 IF INKEY(18)<> 0 THEN GOTO 130
140 INK 0,9:INK 1,6
150 INK 2,0:INK 3,26
160 SYMBOL AFTER 250
170 SYMBOL 255,&0,&0,&0,&0,&39,&3F,&30,&
0
180 SYMBOL 254,&0,&18,&3C,&FF,&FF,&FF,&0
,&0
190 SYMBOL 253,&0,&0,&0,&0,&9C,&F8,&1C,&
0
```

```

200 CLS
210 LET B$=CHR$(143)+"          "+CHR$(143)
220 LET Y=20:LET COL=1
230 LET A$=CHR$(255)+CHR$(254)+CHR$(253)
240 LET X=20
250 WHILE COUNT < 500
260 IF (JOY(0)=4 OR INKEY(64)=0) AND X>1
    THEN LET X=X-1
270 IF (JOY(0)=8 OR INKEY(32)=0) AND X<3
    9 THEN LET X=X+1
280 IF TEST((X*16)-16,400-17)<>0 AND COU
    NT <> 498 AND COUNT <> 598 AND COUNT <> 6
    98 THEN GOTO 560
290 IF TEST((X*16),400-17)<>0 AND COUNT
    <> 498 AND COUNT <> 598 AND COUNT <> 698
    THEN GOTO 560
300 IF TEST((X+1)*16,400-17)<>0 AND COUN
    T <> 498 AND COUNT <> 598 AND COUNT <> 69
    8 THEN GOTO 560
310 IF COUNT=476 OR COUNT=576 OR COUNT=6
    76 THEN LET B$="---FINISH---"
320 LOCATE X,1:PEN 1:PRINT A$
330 LET Y=Y+((INT(RND*2)+1)-(INT(RND*2)+
    1))
340 IF Y<1 THEN LET Y=1
350 IF Y>30 THEN LET Y=30
360 LOCATE y,24:LET COL=3-COL:IF COL=1 T
    HEN PEN 2 ELSE PEN 3
370 PRINT B$
380 LOCATE 40,25:PRINT "  "
390 IF COUNT =476 OR COUNT=576 OR COUNT=
    676 THEN LET B$=CHR$(143)+"          "+CHR$(
    143)
400 LET COUNT=COUNT+1
410 WEND
420 LET FLAG=FLAG+1
430 IF FLAG=2 THEN GOTO 490
440 IF FLAG=3 THEN GOTO 530
450 CLS:PRINT "WELL DONE YOU ARE FORMULA
    3 CHAMPION": LOCATE 7,25:PRINT "PRESS <
    ENTER> TO CONTINUE"
460 IF INKEY(18)<>0 THEN GOTO 460
470 CLS
480 LET COUNT=-100:LET B$=CHR$(143)+"
    "+CHR$(143):GOTO 250

```

```

490 CLS:PRINT "WELL DONE YOU ARE FORMULA
  2 CHAMPION": LOCATE 7,25:PRINT "PRESS <
ENTER> TO CONTINUE"
500 IF INKEY(18)<>0 THEN GOTO 500
510 CLS
520 LET COUNT=-200:LET B$=CHR$(143)+"
  "+CHR$(143):GOTO 250
530 CLS :PRINT "WELL DONE YOU ARE FORMUL
A 1 CHAMPION": LOCATE 7,25:PRINT "PRESS
<ENTER> TO PLAY AGAIN"
540 IF INKEY(18)<>0 THEN GOTO 620
550 RUN
560 LOCATE X,1:PEN 1:PRINT CHR$(238);CHR
$(238);CHR$(238)
570 ENV 1,1,0,50,14,-1,10:SOUND 1,0,0,15
,1,0,31:SOUND 2,0,0,15,1,0,15:SOUND 4,0,
0,15,1,0,1
580 FOR I=1 TO 2000:NEXT
590 PEN 2
600 CLS:LOCATE 1,1:PRINT "*****
*HARD LUCK*****
YOU CRASHED!"
610 LOCATE 7,25:PRINT "PRESS <ENTER> TO
PLAY AGAIN"
620 IF INKEY(18)<>0 THEN GOTO 620
630 RUN 200

```

MINEFIELD

In this game the screen represents a field containing obstacles in the form of mines, craters, trees and bushes. Your objective is to get your man from the bottom line of the screen to the top line without walking into any of these hazards.

You'll have to plan ahead to do this as you must give your man a list of ten instructions to carry out. He does this automatically once you have typed in your commands. Pick from these to map the path:

LLeft
RRight
UUp
DDown

pressing **ENTER** after each letter, and finally **ENTER** to set the man in motion.

Detailed description

30-40 Set up the ink colours and set border, paper and pen colours for printing the instructions.

50-100 Clear the screen. Print the instructions for playing the game.

110-120 Set the variable **Y** to start the player on the bottom line of the screen and dimension the array for storing the moves.

130-170 Install the user-defined graphics characters.

180-190 Clear the screen and set the border colour.

200-260 Go to the sub-routine for printing the obstacles and repeat for 50 obstacles.

270 Select a random position on the bottom line for the player to start. Print the top line on the screen to represent 'home'.

280-320 Invite the player to enter a move. Convert to upper case and check that the entry is valid. Repeat for ten moves.

330-380 Loop for processing the ten moves, calculating the X and Y co-ordinates, and printing a dot when the man moves on to the next position.
390-410 Check that the man has not gone off the screen.
420 Increment each move.
430 Check if the man has run into an obstacle.
440 Print the man on the screen in the current position.
450-470 Generate the sound of the man moving. Return to the loop after a short pause.
480 Return for the next set of ten moves.
490-500 Sub-routine for selecting at random the positions of the obstacles.
510 Check if the player has reached home safely and if so go to the winning routine.
520-550 If the player has run into an obstacle, print a 'splash' and generate the appropriate sound.
560-600 Clear the screen. Print a message and the number of moves taken. Invite the player to play again.
610 Generates a tune if the player has won.
620-640 Print the man waving his arms on the 'home' line.
650-700 Clear the screen. Print a congratulations message and the number of moves taken. Invite the player to have another game.

```

10 REM MINEFIELD
20 REM BY T.BANKS
30 INK 0,9:INK 1,0:INK 2,26:INK 3,21
40 BORDER 9:PAPER 0:PEN 1
50 CLS:LOCATE 16,1:PRINT "MINEFIELD":LOC
ATE 1,3:PRINT "YOU ARE IN A FIELD CONTAI
NING OBSTACLES IN THE FORM OF MINES, CRA
TERS,TREES AND BUSHES."
60 PRINT:PRINT"YOUR OBJECT IS TO GET ACR
OSS THE FIELD WITHOUT WALKING INTO ANY
OF THESE HAZARDS."
70 PRINT:PRINT"YOU MUST ENTER TEN MOVES
IN ADVANCE."
80 PRINT:PRINT"TYPE IN EACH MOVE FOLLOWE
D BY <ENTER>." :PRINT:PRINT" MOVES ARE
: L-LEFT
  
```

```

R-RIGHT
U-UP
D-DOWN"
90 LOCATE 5,25:PRINT "PRESS <ENTER> TO S
TART"
100 IF INKEY(18)<>0 THEN GOTO 100
110 LET Y=23
120 DIM A$(200)
130 SYMBOL AFTER 250
140 SYMBOL 255,&0,&4,&48,&30,&14,&58,&30
,&10
150 SYMBOL 254,&42,&C3,&24,&18,&18,&24,&
C3,&42
160 SYMBOL 253,&38,&7E,&FE,&FE,&7C,&38,&
10,&10
170 SYMBOL 252,&3C,&44,&82,&99,&99,&82,&
44,&3C
180 CLS
190 BORDER 9
200 FOR T=1 TO 50
210 GOSUB 490
220 PEN 3:PRINT CHR$(255)
230 GOSUB 490:PEN 3:PRINT CHR$(253)
240 GOSUB 490:PEN 2:PRINT CHR$(254)
250 GOSUB 490:PEN 2:PRINT CHR$(252)
260 NEXT
270 LET X=INT(RND*40)+1:LOCATE X,23:PEN
1:PRINT CHR$(248):PEN 2:LOCATE 1,1:PRINT
"*****"
**"
280 FOR A=1 TO 10
290 LOCATE 1,24:PRINT "
":LOCATE 1,24
300 PEN 2:PRINT "MOVE ";A;:INPUT " ";A$(
A)
310 IF UPPER$(a$(a)) <> "L" AND UPPER$(A
$(A))<>"R" AND UPPER$(A$(A))<> "U" AND U
PPER$(a$(a))<>"D" THEN GOTO 290
320 NEXT A
330 FOR G=1 TO 10
340 LOCATE X,Y:PRINT "."
350 IF A$(G)="L" OR a$(g)="l" THEN LET X
=X-1
360 IF A$(G)="R" OR a$(g)="r" THEN LET X
=X+1

```

```

370 IF A$(G)="U" OR a$(g)="u" THEN LET Y
=Y-1
380 IF A$(G)="D" OR a$(g)="d" THEN LET Y
=Y+1
390 IF X<1 THEN LET X=1
400 IF X>40 THEN LET X=40
410 IF Y>23 THEN LET Y=23
420 LET MOVES=MOVES+1
430 IF TEST(X*16-9,408-(Y*16))<>0 THEN G
OTO 510
440 LOCATE X,Y:PEN 1:PRINT CHR$(248)
450 SOUND 1,3822,5,7
460 FOR F=1 TO 1000:NEXT F
470 NEXT
480 GOTO 280
490 LOCATE INT(RND*40)+1,INT(RND*21)+2
500 RETURN
510 IF Y=1 THEN GOTO 610
520 ENT 1,30,20,5
530 SOUND 3,300,0,15,15,1,0
540 LOCATE X,Y:PRINT CHR$(238)
550 FOR I=1 TO 2000 :NEXT
560 CLS
570 LOCATE 1,1:PRINT "HARD LUCK! YOU RAN
    INTO AN OBSTACLE":PRINT:PRINT:PRINT"YOU
    HAD ";MOVES;" MOVES"
580 LOCATE 5,25:PRINT "PRESS <ENTER> TO
    PLAY AGAIN"
590 IF INKEY(18)<>0 THEN GOTO 590
600 RUN 110
610 ENT 1,40,-15,5:SOUND 3,800,200,15,15
    ,1,0:FOR W=1 TO 200
620 LOCATE X,Y:PEN 1:PRINT CHR$(249)
630 LOCATE X,Y:PRINT CHR$(248)
640 NEXT
650 CLS
660 PRINT "CONGRATULATIONS! YOU GOT HOME
    SAFELY"
670 PRINT:PRINT:PRINT"YOU REACHED HOME I
    N ";MOVES;" MOVES"
680 LOCATE 5,25:PRINT "PRESS <ENTER> TO
    PLAY AGAIN"
690 IF INKEY(18)<>0 THEN GOTO 690
700 RUN 110

```



Splash is a complex new board game for you to play against the computer. It is played on a seven by seven board and the object of the game is to capture more squares than the computer. Thus, the first player to capture 25 or more squares is the winner. You and the computer take it in turns to play in any free square, but playing in a square also affects those surrounding it.

Initially, all squares are black. As the game progresses they change colour according to the scheme: RED - GREEN - YELLOW - BLUE - WHITE. If a square is left white at the end of a move it is captured and initialled Y(You) or M(Me) - the Amstrad). Playing in a square increases its colour by three steps. The eight squares immediately surrounding it have their colours increased by two steps, and the sixteen squares around these are increased by one step. The sequence of colours is displayed at the side of the screen to help you. You enter your move by giving your chosen square's grid reference, a letter and number.

Detailed description

30 Set the computer in mode 1 and set up the colours for printing the instructions. Clear the screen.

40-110 Print the instructions for playing the game.

120-160 Install the user-defined graphics characters.

170-190 Put the computer in mode 0 as the game will

require more than the four colours available in mode 1. Set the ink colours for the game.

200 Dimension the array for the board.

210-220 Set the pen, border and paper colours for printing the initial screen.

230-370 Print the screen display.

380-390 Ask the player to input a move, and convert the entry to upper case.

400-460 Check the move that has been entered to ensure it is valid.

470-480 Check the total number of captured squares and if it is 49 then go to the end of game routine.

490-610 Reset the variables for the computer's move and print a message on the screen. Go to the routine for calculating the best move. Check for end of game, then return for the player's next move if the game is not over.

620-940 Check the move and go to the routine to print the new colours on the screen.

950-1070 Routine for working out the computer's best move.

1080-1150 Return the computer to mode 1. Print the scores and the end of game message. Invite the player to have another game.

1160-1220 Print the new colours on the screen after a move has been made and return for the next move.

```
10 REM SPLASH
20 REM T.BANKS
30 MODE 1:INK 0,0:INK 1,14:BORDER 14:PAPER 1:PEN 0:CLS
40 LOCATE 18,1:PRINT "SPLASH!":PRINT:PRINT"THIS IS A NEW BOARD GAME TO PLAY AGAINSTTHE COMPUTER. IT IS PLAYED ON A 7 BY 7 BOARD AND THE OBJECT OF THE GAME IS TO CAPTURE MORE SQUARES THAN THE COMPUTER."
50 PRINT:PRINT "THUS, THE FIRST PLAYER TO CAPTURE 25 OR MORE SQUARES IS THE WINNER.":PRINT:PRINT "WE TAKE IT IN TURNS TO PLAY IN ANY FREE SQUARE - PLAYING IN A
```

```

SQUARE AFFECTS      THOSE SURROUNDING IT."
60 PRINT:PRINT "INITIALLY, ALL SQUARES ARE BLACK. AS THE GAME PROGRESSES THEY CHANGE COLOUR      ACCORDING TO THE SCHEME:
":PRINT:PRINT"RED - GREEN - YELLOW - BLUE - WHITE."
70 PRINT:PRINT "IF A SQUARE IS LEFT WHITE AT THE END OF A MOVE IT IS CAPTURED AND INITIALLED      (Y=YOU M=ME). "
80 LOCATE 5,25:PRINT "PRESS ANY KEY TO CONTINUE":IF INKEY$="" THEN GOTO 80
90 CLS:LOCATE 18,1:PRINT "SPLASH!":PRINT:PRINT "PLAYING IN A SQUARE INCREASES ITS COLOUR BY THREE STEPS (THE SEQUENCE OF COLOURS IS GIVEN AT THE SIDE OF THE BOARD TO      HELP YOU). "
100 PRINT:PRINT"THE 8 SQUARES IMMEDIATELY SURROUNDING IT HAVE THEIR COLOURS INCREASED BY TWO      STEPS AND THE 16 SQUARES AROUND THESE      ARE INCREASED BY ONE STEP."
105 PRINT:PRINT "ENTER YOUR MOVE AS A LETTER FOLLOWED BY A DIGIT.":PRINT:PRINT "TO QUIT PRESS Q WHEN ASKED FOR YOUR MOVE
"
110 LOCATE 5,25:PRINT "PRESS ANY KEY TO CONTINUE":IF INKEY$="" THEN GOTO 110
120 SYMBOL AFTER 240
130 SYMBOL 255,&0,&0,&0,&0,&C6,&EE,&FE,&FE
140 SYMBOL 254,&D6,&C6,&C6,&0,&0,&0,&0,&0
150 SYMBOL 253,&0,&0,&0,&0,&66,&66,&66,&3C
160 SYMBOL 252,&18,&18,&3C,&0,&0,&0,&0,&0
170 MODE 0
180 INK 0,0:INK 1,6:INK 2,18:INK 3,24:INK 4,1:INK 5,26:INK 6,14
190 CLS
200 DIM BOARD(7,7)
210 PEN 0
220 BORDER 14:PAPER 6:CLS
230 LOCATE 8,1:PRINT"SPLASH"
240 FOR A=1 TO 7

```

```

250 LET A$=A$+CHR$(143)
260 NEXT
270 FOR A1=6 TO 19
280 LOCATE 8,A1:PRINT A$
290 NEXT
300 LOCATE 8,4:PRINT "ABCDEFGG"
310 FOR A2=6 TO 19 STEP 2
320 LOCATE 5,A2:LET CO=CO+1:PRINT CO
330 NEXT
340 FOR A3=6 TO 16 STEP 2
350 LET COUNT=COUNT+1
360 LOCATE 19,A3:PEN COUNT:PRINT CHR$(14
3)
370 NEXT
375 PEN 5:LOCATE 1,21:PRINT "ENTER: LETT
ER,DIGIT"
380 PEN 0: LOCATE 1,25:PRINT "YOU ";YOU;
" "; "ME ";ME:LOCATE 1,23:PRINT "
":LOCATE
1,23:INPUT "YOUR MOVE ";B$
390 LET B$=UPPER$(B$)
400 IF UPPER$(B$)="Q" THEN GOTO 1080
410 LET X=ASC(LEFT$(B$,1))-64
420 IF X<1 OR X>7 THEN GOTO 380
430 LET Y=ASC(RIGHT$(B$,1))-48
440 IF Y<1 OR Y>7 THEN GOTO 380
450 IF BOARD(X,Y)=5 THEN GOTO 380
460 LET FLAG=0
470 IF YOU+ME=49 THEN GOTO 1080
480 GOSUB 620:REM CHECK MOVE
490 REM COMPUTER MOVE
500 LET J=0:LET HIGH=0:LET CX=0:LET CY=0
510 LET FLAG=1
520 LOCATE 1,23:PRINT "
":PEN 0:LOCATE 1,23
:PRINT "I AM THINKING":LOCATE 1,25:PRINT
"YOU ";YOU;" "; "ME ";ME
530 FOR B=1 TO 7
540 FOR D=1 TO 7
550 IF BOARD(D,B)=5 THEN GOTO 570
560 GOSUB 950
570 NEXT
580 NEXT
590 LET X=CX:LET Y=CY:GOSUB 620
600 IF YOU+ME=49 THEN GOTO 1080

```

```

610 GOTO 380
620 REM CHECK MOVE
630 LET L1=X:LET L2=Y:LET C=3:GOSUB 1160
640 IF Y-1 < 1 THEN GOTO 680
650 IF X-1>0 THEN LET L1=X-1:L2=Y-1:C=2:
GOSUB 1160
660 LET L1=X:LET L2=Y-1:LET C=2:GOSUB 11
60
670 IF X+1 < 8 THEN LET L1=X+1:L2=Y-1:C=2
:GOSUB 1160
680 IF Y+1 > 7 THEN GOTO 720
690 IF X-1>0 THEN LET L1=X-1:LET L2=Y+1:
C=2:GOSUB 1160
700 LET L1=X:L2=Y+1:C=2:GOSUB 1160
710 IF X+1<8 THEN LET L1=X+1:L2=Y+1:C=2:
GOSUB 1160
720 IF X-1>0 THEN LET L1=X-1:L2=Y:C=2:GO
SUB 1160
730 IF X+1<8 THEN LET L1=X+1:L2=Y:C=2:GO
SUB 1160
740 IF Y-2<1 THEN GOTO 800
750 IF X-2>0 THEN L1=X-2:L2=Y-2:C=1:GOSU
B 1160
760 IF X-1>0 THEN L1=X-1:L2=Y-2:C=1:GOSU
B 1160
770 L1=X:L2=Y-2:C=1:GOSUB 1160
780 IF X+1<8 THEN L1=X+1:L2=Y-2:C=1:GOSU
B 1160
790 IF X+2<8 THEN L1=X+2:L2=Y-2:C=1:GOSU
B 1160
800 IF Y+2>7 THEN GOTO 860
810 IF X-2>0 THEN L1=X-2:L2=Y+2:C=1:GOSU
B 1160
820 IF X-1>0 THEN L1=X-1:L2=Y+2:C=1:GOSU
B 1160
830 L1=X:L2=Y+2:C=1:GOSUB 1160
840 IF X+1<8 THEN L1=X+1:L2=Y+2:C=1:GOSU
B 1160
850 IF X+2<8 THEN L1=X+2:L2=Y+2:C=1:GOSU
B 1160
860 IF X-2<1 THEN GOTO 900
870 IF Y+1<8 THEN L1=X-2:L2=Y+1:C=1:GOSU
B 1160
880 L1=X-2:L2=Y:C=1:GOSUB 1160

```

```

890 IF Y-1>0 THEN L1=X-2:L2=Y-1:C=1:GOSU
B 1160
900 IF X+2>7 THEN GOTO 940
910 IF Y-1>0 THEN L1=X+2:L2=Y-1:C=1:GOSU
B 1160
920 L1=X+2:L2=Y:C=1:GOSUB 1160
930 IF Y+1<8 THEN L1=X+2:L2=Y+1:C=1:GOSU
B 1160
940 RETURN
950 REM COMPUTER CHECK
960 IF BOARD(D,B)+3=5 THEN LET J=J+200 E
LSE LET J=J+BOARD(D,B)
970 IF D+2<8 THEN IF BOARD(D+2,B)+1=5 TH
EN LET J=J+200 ELSE LET J=J+BOARD(D+2,B)
980 IF D+1<8 THEN IF BOARD(D+1,B)+2=5 TH
EN LET J=J+200 ELSE LET J=J+BOARD(D+1,B)
990 IF D-2>0 THEN IF BOARD(D-2,B)+1=5 TH
EN LET J=J+200 ELSE LET J=J+BOARD(D-2,B)

1000 IF D-1>0 THEN IF BOARD(D-1,B)+2=5 T
HEN LET J=J+200 ELSE LET J=J+BOARD(D-1,B
)
1010 IF B-1>0 THEN IF BOARD(D,B-1)+2=5 T
HEN LET J=J+200 ELSE LET J=J+BOARD(D,B-1
)
1020 IF B-2>0 THEN IF BOARD(D,B-2)+1=5 T
HEN LET J=J+200 ELSE LET J=J+BOARD(D,B-2
)
1030 IF B+1<8 THEN IF BOARD(D,B+1)+2=5 T
HEN LET J=J+200 ELSE LET J=J+BOARD(D,B+1
)
1040 IF B+2<8 THEN IF BOARD(D,B+2)+1=5 T
HEN LET J=J+200 ELSE LET J=J+BOARD(D,B+2
)
1050 IF J>HIGH THEN LET HIGH=J:LET CX=D:
LET CY=B:LET J=0:RETURN
1060 LET J=0
1070 RETURN
1080 MODE 1:CLS:IF YOU<ME THEN GOTO 1140

1090 PRINT "CONGRATULATIONS YOU BEAT ME
"
1100 PRINT:PRINT"YOU SCORED ";YOU:PRINT:
PRINT"I SCORED ";ME

```

```

1110 LOCATE 5,25:PRINT "PRESS ANY KEY TO
    START"
1120 IF INKEY$="" THEN GOTO 1120
1130 RUN
1140 PRINT "I WON EASILY ":GOTO 1100
1150 END
1160 REM PRINT MOVE AND UPDATE
1170 IF BOARD(L1,L2)=5 THEN RETURN
1180 LET BOARD(L1,L2)=BOARD(L1,L2)+C
1190 IF BOARD(L1,L2)>5 THEN LET BOARD(L1
    ,L2)=BOARD(L1,L2)-5
1200 LOCATE L1+7,(L2*2)+4:PEN BOARD(L1,L
    2):PRINT CHR$(143):LOCATE L1+7,(L2*2)+5:
    PRINT CHR$(143)
1210 IF BOARD(L1,L2)=5 THEN IF FLAG=1 TH
    EN LOCATE L1+7,(L2*2)+4:PEN BOARD(L1,L2)
    :PRINT CHR$(255):LOCATE L1+7,(L2*2)+5:PR
    INT CHR$(254):LET ME=ME+1:ELSE LOCATE L1
    +7,(L2*2)+4:PEN BOARD(L1,L2):PRINT CHR$(
    253):LOCATE L1+7,(L2*2)+5:PRINT CHR$(252
    ):LET YOU=YOU+1
1220 RETURN

```

PONTOON

This is the traditional card game, rewritten for the Amstrad so that you can test your skill and luck against the computer. The aim is to get a hand of five or fewer cards which add up to 21; or as near to that number as possible.

The cards score as follows; Court cards score 10 each, an Ace scores 1 or 11 and the rest score their face values. If you get more than 21, you go 'Bust'. You are dealt two cards and then have the choice of Stick or Twist, by typing S or T. If you Stick the Amstrad is dealt more cards and your hand is left unchanged. If you Twist you get more cards, one at a time. You can only Stick if your hand is worth 16 or more.

Detailed description

30 Set the ink, border, paper and pen colours.

40-70 Clear the screen and print the instructions for playing the game.

80 Clear the screen and ask the player to enter his or her name.

90 If player's name is not entered use 'human' as default name!

100-120 Clear the screen. Dimension the arrays and initialize counts.

130-250 Install the user-defined graphics characters.

260 Reset the colours.

270-300 Initialize flags and set variables to zero.

310 Go to the sub-routine for printing the initial hands of cards.

320 Start of main control loop for the player's hand.

330-340 Print the players' names and scores.

350 Go to the sub-routine for determining the value and suit of a random card.

360 Increment the card count.

370 Go to the routine appropriate to the number of cards held in the hand.

380-410 Check for the total value of the cards held in the hand and set the appropriate flags.

420-460 Ask the player to enter either S (Stick) or T (Twist). Check that the input is valid and set the appropriate flags.

470 Return to the start of the loop.

480 Check for Aces in the player's hand.

490 Go to the sub-routine in line 790.

500 Check if the hand has exceeded 21 points. If so print 'Bust', increment the score and return to deal next hand.

510 Print 'Stick'.

520-560 Turn the bank's concealed card over so that it may be seen. Determine the next course of action and go to the appropriate sub-routine.

570-620 Repeat the above for the bank's third card.

630-660 Repeat for the fourth card.

670-700 Repeat for the fifth card.

710 Check to determine which player has won.

720 Print message for the bank winning.

730-740 Print message for the player having a five card trick.

750-760 Print message for the player having won with the higher score.

770-780 Print message for the bank having won with a five card trick.

790 Clear the bottom line.

800-880 Determine the value and suit of a random card.

890-1040 Routine for printing the cards on the screen.

1050-1180 Sub-routine for printing the initial hands of cards.

1190-1230 Select the graphics appropriate to the suit of the card.

1240-1310 Sub-routine for determining the value of the card.

1320-1360 Print the card on the screen.

1370-1380 Calculation for Aces held in the computer's hand.

```

10 REM PONTOON
20 REM BY T.BANKS
30 INK 0,9:INK 1,0:INK 2,6:INK 3,26:BORD
ER 9:PAPER 0:PEN 1
40 CLS:LOCATE 18,1:PRINT"PONTOON":PRINT:
PRINT"THE OBJECT OF THE GAME IS TO GET A
S      CLOSE TO A VALUE OF 21 AS POSSIBLE
BY"
45 PRINT "ADDING THE FACE VALUES OF THE
CARDS IN YOUR HAND, AND THEN FOR THE C
OMPUTER TOATTEMPT TO MATCH THIS, OR GET
CLOSER      STILL WITHOUT EXCEEDING 21 AND
SO GOING 'BUST'."
50 PRINT:PRINT"COURT CARDS COUNT 10 AND
THE ACE HAS A VALUE OF EITHER 1 OR 11."
:PRINT:PRINT"T- DEALS THE PLAYER ANOTHER
CARD      S- FINISHES THE PLAYERS HAN
D AND DEALS      THE COMPUTER"
60 PRINT:PRINT "PRESS ANY KEY TO DEAL NE
W HAND."
70 LOCATE 5,25:PRINT "PRESS ANY KEY TO C
ONTINUE":IF INKEY$="" THEN GOTO 70
80 CLS: INPUT "PLEASE TYPE IN YOUR NAME
";NAME$
90 IF NAME$="" THEN LET NAME$="HUMAN"
100 CLS
110 DIM C$(4):DIM PACK(52)
120 LET AMSTRAD=0:LET NAME=0
130 SYMBOL AFTER 240
140 SYMBOL 255,&0,&3,&7,&7,&7,&7,&3,&3D
150 SYMBOL 254,&0,&C0,&E0,&E0,&E0,&E0,&C
0,&BC
160 SYMBOL 253,&7F,&7F,&7F,&7F,&3D,&1,&3
,&3
170 SYMBOL 252,&FE,&FE,&FE,&FE,&BC,&80,&
C0,&C0
180 SYMBOL 251,&1C,&3E,&7E,&FF,&FF,&FF,&
FF,&FF
190 SYMBOL 250,&70,&F8,&FC,&FE,&FE,&FE,&
FE,&FE
200 SYMBOL 249,&FF,&7F,&3F,&1F,&F,&7,&3,
&1
210 SYMBOL 248,&FE,&FC,&F8,&F0,&E0,&C0,&
80,&0

```

```

220 SYMBOL 247,&1,&3,&7,&F,&1F,&3F,&7F,&
7F
230 SYMBOL 246,&80,&C0,&E0,&F0,&F8,&FC,&
FE,&FE
240 SYMBOL 245,&7F,&7F,&7F,&3D,&19,&1,&3
,&3
250 SYMBOL 244,&FE,&FE,&FE,&BC,&98,&80,&
C0,&C0
260 PAPER 0:CLS:BORDER 9
270 LET ACE=0:LET CACE=0:CLS
280 LET FLAG=0:LET VALUE=0:LET BANK=0
290 FOR Q=1 TO 52:LET PACK(Q)=0:NEXT:FOR
Q1=1 TO 4:LET C$(4)="":NEXT
300 LET COUNT=0
310 GOSUB 1050
320 WHILE FLAG=0
330 LOCATE 2,11:PRINT NAME$;" ";HUMAN
340 LOCATE 2,13:PRINT "AMSTRAD ";AMSTRAD
345 PAPER 0:BORDER 9
350 GOSUB 800
360 LET COUNT=COUNT+1
370 ON COUNT GOSUB 890,930,960,990,1020
380 IF CARD=1 THEN LET ACE=1
390 IF COUNT=1 THEN GOTO 350
400 IF VALUE>21 THEN LET FLAG=2:GOTO 470
410 IF COUNT=5 THEN LET FLAG=3:LET FCT=1
420 LOCATE 1,25:PEN 1:PRINT "(T)WIST, (S
)TICK "
430 LET A$=INKEY$:IF A$<>"T" AND A$<>"t"
AND A$<>"s" AND A$<>"S" THEN GOTO 430
440 IF UPPER$(A$)="S" AND VALUE<16 AND A
CE=0 THEN GOTO 430
450 IF UPPER$(A$)="S" AND ACE=1 AND VAL
UE+10<16 THEN GOTO 430
460 IF UPPER$(A$)="S" THEN LET FLAG=1
470 WEND
480 IF ACE=1 AND VALUE+10<22 THEN LET VA
LUE=VALUE+10:LET ACE=0
490 GOSUB 790
500 IF FLAG=2 THEN LOCATE 1,25:PRINT "BU
ST !! ":IF INKEY$="" THEN GOTO 500:ELSE
LET AMSTRAD=AMSTRAD+1:GOTO 260
510 LOCATE 35,5:PEN 2:PRINT "STICK"
520 GOSUB 800:GOSUB 1190:GOSUB 1240:LET
BANK=BANK+CARD

```

```

530 IF CARD=1 THEN LET CACE=1
540 LOCATE 10,15:PAPER 3:PEN COL:GOSUB 1
320:LOCATE 10,16:GOSUB 1330:LOCATE 10,17
:GOSUB 1330:LOCATE 10,18:GOSUB 1340:LOCA
TE 10,19:GOSUB 1350:LOCATE 10,20:GOSUB 1
330:LOCATE 10,21:GOSUB 1330:LOCATE 10,22
:GOSUB 1360
550 GOSUB 1370
560 IF BANK > 16 THEN GOSUB 790:GOTO 710
570 GOSUB 800:GOSUB 1190:GOSUB 1240:LET
BANK=BANK+CARD
580 IF CARD=1 THEN LET CACE=1
590 LOCATE 17,15:PAPER 3:PEN COL:GOSUB 1
320:LOCATE 17,16:GOSUB 1330:LOCATE 17,17
:GOSUB 1330:LOCATE 17,18:GOSUB 1340:LOCA
TE 17,19:GOSUB 1350:LOCATE 17,20:GOSUB 1
330:LOCATE 17,21:GOSUB 1330:LOCATE 17,22
:GOSUB 1360
600 GOSUB 1370
610 IF BANK> 16 AND BANK<22 THEN GOTO 71
0
620 IF BANK>21 THEN GOTO 760
630 GOSUB 800:GOSUB 1190:GOSUB 1240:LET
BANK=BANK+CARD
640 LOCATE 24,15:PAPER 3:PEN COL:GOSUB 1
320:LOCATE 24,16:GOSUB 1330:LOCATE 24,17
:GOSUB 1330:LOCATE 24,18:GOSUB 1340:LOCA
TE 24,19:GOSUB 1350:LOCATE 24,20:GOSUB 1
330:LOCATE 24,21:GOSUB 1330:LOCATE 24,22
:GOSUB 1360
650 IF BANK> 16 AND BANK<22 THEN GOTO 71
0
660 IF BANK>21 THEN GOTO 760
670 GOSUB 800:GOSUB 1190:GOSUB 1240:LET
BANK=BANK+CARD
680 LOCATE 31,15:PAPER 3:PEN COL:GOSUB 1
320:LOCATE 31,16:GOSUB 1330:LOCATE 31,17
:GOSUB 1330:LOCATE 31,18:GOSUB 1340:LOCA
TE 31,19:GOSUB 1350:LOCATE 31,20:GOSUB 1
330:LOCATE 31,21:GOSUB 1330:LOCATE 31,22
:GOSUB 1360
690 IF BANK>21 THEN GOTO 760
700 GOTO 770
710 GOSUB 790:IF FLAG=3 THEN GOTO 730
720 IF BANK >= VALUE THEN LOCATE 1,25:PR

```

```

INT "THE BANK WINS WITH ";BANK:IF INKEY$="" THEN GOTO 720:ELSE LET AMSTRAD=AMSTRAD+1:GOTO 260
730 IF FCT=0 THEN GOTO 750 ELSE LOCATE 1,25:PRINT "WELL DONE YOU WIN WITH A FIVE CARD TRICK"
740 IF INKEY$="" THEN GOTO 740 ELSE LET FCT=0:HUMAN=HUMAN+1:GOTO 260
750 LOCATE 1,25:PRINT "WELL DONE YOU WIN WITH A SCORE OF ";VALUE:IF INKEY$="" THEN GOTO 750:ELSE LET HUMAN=HUMAN+1:GOTO 260
760 GOTO 730
770 GOSUB 790
780 LOCATE 1,25:PRINT "THE BANK WINS WITH A 5 CARD TRICK":IF INKEY$="" THEN GOTO 780:ELSE LET AMSTRAD=AMSTRAD+1:GOTO 260
790 LOCATE 1,25:PRINT "
                                ":RETURN
800 LET CARD=INT(RND*52)+1
810 IF PACK(CARD)=1 THEN GOTO 800
820 LET PACK(CARD)=1
830 IF CARD<27 THEN LET COL=1 ELSE LET COL=2
840 IF CARD<14 THEN LET SUIT$="SPADES"
850 IF CARD>13 AND CARD<27 THEN LET SUIT$="CLUBS":LET CARD=CARD-13
860 IF CARD>26 AND CARD<40 THEN LET SUIT$="DIAMONDS":LET CARD=CARD-26
870 IF CARD>39 AND CARD<53 THEN LET SUIT$="HEARTS":LET CARD=CARD-39
880 RETURN
890 GOSUB 1190:GOSUB 1240:LOCATE 2,2:PAPER 3:PEN COL:GOSUB 1320:LOCATE 2,3:GOSUB 1330:LOCATE 2,4:GOSUB 1330:LOCATE 2,5:GOSUB 1340:LOCATE 2,6:GOSUB 1350
900 LOCATE 2,7:GOSUB 1330:LOCATE 2,8:GOSUB 1330:LOCATE 2,9:GOSUB 1360
910 LET VALUE=VALUE+CARD
920 RETURN
930 GOSUB 1190:GOSUB 1240:LOCATE 10,2:PAPER 3:PEN COL:GOSUB 1320:LOCATE 10,3:GOSUB 1330:LOCATE 10,4:GOSUB 1330:LOCATE 10,5:GOSUB 1340:LOCATE 10,6:GOSUB 1350:LOC

```

```

ATE 10,7:GOSUB 1330:LOCATE 10,8:GOSUB 13
30:LOCATE 10,9:GOSUB 1360
940 LET VALUE=VALUE+CARD
950 RETURN
960 GOSUB 1190:GOSUB 1240:LOCATE 17,2:PA
PER 3:PEN COL:GOSUB 1320:LOCATE 17,3:GOS
UB 1330:LOCATE 17,4:GOSUB 1330:LOCATE 17
,5:GOSUB 1340:LOCATE 17,6:GOSUB 1350:LOC
ATE 17,7:GOSUB 1330:LOCATE 17,8:GOSUB 13
30:LOCATE 17,9:GOSUB 1360
970 LET VALUE=VALUE+CARD
980 RETURN
990 GOSUB 1190:GOSUB 1240:LOCATE 24,2:PA
PER 3:PEN COL:GOSUB 1320:LOCATE 24,3:GOS
UB 1330:LOCATE 24,4:GOSUB 1330:LOCATE 24
,5:GOSUB 1340:LOCATE 24,6:GOSUB 1350:LOC
ATE 24,7:GOSUB 1330:LOCATE 24,8:GOSUB 13
30:LOCATE 24,9:GOSUB 1360
1000 LET VALUE=VALUE+CARD
1010 RETURN
1020 GOSUB 1190:GOSUB 1240:LOCATE 31,2:P
APER 3:PEN COL:GOSUB 1320:LOCATE 31,3:GO
SUB 1330:LOCATE 31,4:GOSUB 1330:LOCATE 3
1,5:GOSUB 1340:LOCATE 31,6:GOSUB 1350:LO
CATE 31,7:GOSUB 1330:LOCATE 31,8:GOSUB 1
330:LOCATE 31,9:GOSUB 1360
1030 LET VALUE=VALUE+CARD
1040 RETURN
1050 GOSUB 800
1060 GOSUB 1190
1070 GOSUB 1240
1080 LET BANK=BANK+CARD
1090 LOCATE 2,15:PAPER 3:PEN COL:GOSUB 1
320
1100 LOCATE 2,16:GOSUB 1330:LOCATE 2,17:
GOSUB 1330
1110 LOCATE 2,18:GOSUB 1340:LOCATE 2,19:
GOSUB 1350
1120 LOCATE 2,20:GOSUB 1330:LOCATE 2,21:
GOSUB 1330:LOCATE 2,22:GOSUB 1360
1130 FOR U=1 TO 8
1140 PEN 2
1150 LOCATE 10,U+14
1160 FOR I=1 TO 6:PRINT CHR$(207);:NEXT:
PRINT

```

```

1170 NEXT
1180 RETURN
1190 IF SUIT$="SPADES" THEN LET C$(1)=CHR$(247):C$(2)=CHR$(246):C$(3)=CHR$(245):C$(4)=CHR$(244)
1200 IF SUIT$="HEARTS" THEN LET C$(1)=CHR$(251):C$(2)=CHR$(250):C$(3)=CHR$(249):C$(4)=CHR$(248)
1210 IF SUIT$="CLUBS" THEN LET C$(1)=CHR$(255):C$(2)=CHR$(254):C$(3)=CHR$(253):C$(4)=CHR$(252)
1220 IF SUIT$="DIAMONDS" THEN LET C$(1)=CHR$(214):C$(2)=CHR$(215):C$(3)=CHR$(213):C$(4)=CHR$(212)
1230 RETURN
1240 IF CARD>1 AND CARD<10 THEN LET Y$=STR$(CARD):LET Y$=RIGHT$(Y$,1)
1250 IF CARD=11 THEN LET Y$="J"
1260 IF CARD=12 THEN LET Y$="Q"
1270 IF CARD=13 THEN LET Y$="K"
1280 IF CARD=1 THEN LET Y$="A"
1290 IF CARD=10 THEN LET Y$="T"
1300 IF CARD=11 OR CARD=12 OR CARD=13 THEN LET CARD=10
1310 RETURN
1320 PRINT Y$;CHR$(32);CHR$(32);CHR$(32);CHR$(32);CHR$(32):RETURN
1330 PRINT CHR$(32);CHR$(32);CHR$(32);CHR$(32);CHR$(32):RETURN
1340 PRINT CHR$(32);CHR$(32);C$(1);C$(2);CHR$(32);CHR$(32):RETURN
1350 PRINT CHR$(32);CHR$(32);C$(3);C$(4);CHR$(32);CHR$(32):RETURN
1360 PRINT CHR$(32);CHR$(32);CHR$(32);CHR$(32);CHR$(32);Y$:RETURN
1370 IF CACE=1 AND BANK+10>16 AND BANK+10<22 THEN LET BANK=BANK+10:LET CACE=0:RETURN
1380 RETURN

```

TARGET PRACTICE

This is a target shooting game for two players; player number one having the left-hand side of the screen and player number two the right-hand side. Shots must be taken alternately.

There are ten targets to destroy and you are allowed only fifteen shots in which to do it. If the targets are not all destroyed, the score is expressed as a percentage for each player at the end of the game.

The controls are: player one fires by pressing **A** and player two fires by pressing **L**. You can also choose the speed of your shot from a speedy 1 to a slow 9.

Detailed description

30 Install the user-defined graphics characters.

40 Set the mode and ink, border, paper and pen colours.

50-90 Print the instructions for playing the game.

100-120 Initialize the variables.

130-160 Clear the screen. Invite the players to enter their names. In default of names being entered use 'One' and 'Two'.

170-180 Clear the screen. Invite the players to enter the speed at which they wish to play the game. Check the entry to ensure it is within the permitted range.

190 Clear the screen. Go to the sub-routine for the score.

200-310 Print the screen display. Print the targets at random for both players.

320-330 Start of the loop for the first player to play.

340 Go to the sub-routine at 520.

350 Check to see if the player has destroyed all the targets. If so, go to the end of game routine.

360 End of the loop for the first player.

370 Start of the loop for the second player to play.

380 Go to the sub-routine at 520.

390 Check to see if the player has destroyed all the targets.

If so, go to the end of game routine.

400 End of the loop for the second player.

410-420 Check if all shots have been fired. If so go to the end of game routine. If not loop back to continue the game.

430 Clear the buffer.

440-510 End of game routine. Print the players' averages and the name of the winner. Invite the players to have another game.

520-630 Determine the colour and position of the firing point. Scan the keyboard for an entry to fire.

640-870 Routine to print the missile, and increment the flags and counters.

880-970 Routine to print the updated scores etc.

```
10 REM TARGET PRACTICE
```

```
20 REM T.BANKS
```

```
30 SYMBOL 255,255,129,189,189,189,189,129,255
```

```
40 MODE 1:INK 0,0:INK 1,6:INK 2,26:INK 3,15:BORDER 0:PAPER 0:PEN 2
```

```
50 CLS:LOCATE 14,1:PRINT "TARGET PRACTICE":PRINT:PRINT"THIS IS A GAME FOR TWO PLAYERS.":PRINT:PRINT"THE OBJECT OF THE GAME IS TO DESTROY AS MANY TARGETS AS YOU CAN WITH YOUR 15 SHOTS."
```

```
60 PRINT:PRINT"THE PLAYER ON THE LEFT USES THE LETTER 'A' TO FIRE AND THE PLAYER ON THE RIGHT USES THE LETTER 'L' TO FIRE.":PRINT:PRINT "SHOTS MUST BE FIRED ALTERNATELY."
```

```
70 PRINT:PRINT"THE SPEED OF THE MISSILE LAUNCHER MAY BE VARIED BETWEEN FAST (1) TO SLOW (9)."
```

```
80 LOCATE 5,25:PRINT "PRESS ANY KEY TO START"
```

```
90 IF INKEY$="" THEN GOTO 90
```

```
100 DIM SCREEN(40,25):DIM SCORE(2):DIM SHOT(2):FLAG=1
```

```
110 LET SHOT(1)=15:LET SHOT(2)=15
```

```
120 LET CV=21
```

```
130 CLS:INPUT "PLAYER 1'S NAME ";A$
```

```
140 IF A$="" THEN LET A$="ONE"
```



```

150 PRINT:INPUT "PLAYER 2'S NAME ";B$
160 IF B$="" THEN LET B$="TWO"
170 CLS: PRINT:INPUT "SPEED LEVEL 1-9 ";
SP
180 IF SP<1 OR SP>9 THEN GOTO 170
190 CLS:GOSUB 880
200 LOCATE 2,23:PRINT " _____
"
210 PLOT 320,56:DRAWR 0,272:DRAWR 319,0:
DRAWR -639,0
220 FOR D=1 TO 10
230 LET P1=INT(RND*19)+1:LET P2=INT(RND*
12)+8
240 IF SCREEN(P1,P2)=1 THEN GOTO 230
250 LOCATE P1,P2:PEN 1:PRINT CHR$(255):L
ET SCREEN(P1,P2)=1
260 NEXT
270 FOR E=1 TO 10
280 LET P1=INT(RND*19)+22:LET P2=INT(RND
*12)+8
290 IF SCREEN(P1,P2)=1 THEN GOTO 280
300 LOCATE P1,P2:PEN 2:PRINT CHR$(255):L
ET SCREEN(P1,P2)=1
310 NEXT
320 REM PLAY GAME
330 WHILE FLAG=1
340 GOSUB 520
350 IF SCORE(1)=10 THEN CLS:GOSUB 880:GO
TO 430
360 WEND
370 WHILE FLAG=2
380 GOSUB 520
390 IF SCORE(2)=10 THEN CLS:GOSUB 880:GO
TO 430
400 WEND
410 IF SHOT(2)<>0 THEN GOTO 330
420 CLS:GOSUB 880
430 FOR G=1 TO 255:LET L$=INKEY$:NEXT
440 PRINT:PRINT:PRINT A$;" 'S AVERAGE WAS
";(SCORE(1)/10)*100
450 PRINT:PRINT:PRINT B$;" 'S AVERAGE WAS
";(SCORE(2)/10)*100
460 IF SCORE(1)>SCORE(2) THEN PRINT:PRI
NT:PRINT A$;" WON":GOTO 490

```

```

470 IF SCORE(2)>SCORE(1) THEN PRINT: PRINT:PRINT B$;" WON":GOTO 490
480 PRINT:PRINT:PRINT"IT WAS A DRAW"
490 LOCATE 5,25:PRINT "PRESS ANY KEY TO
PLAY AGAIN"
500 IF INKEY$="" THEN GOTO 500
510 RUN 100
520 LET BA=BA+1
530 FOR P=0 TO (SP-1)*10:NEXT
540 PEN FLAG
550 IF BA=41 THEN LET BA=1:LOCATE 40,23:
PRINT " ":GOTO 590
560 IF BA=1 THEN GOTO 590
570 IF BA=2 THEN LOCATE 1,23:PRINT" ":GO
TO 590
580 PEN 2:LOCATE BA-1,23:PRINT"_":PEN FL
AG
590 LOCATE BA,23:PRINT CHR$(239)
600 IF BA=21 THEN GOTO 630
610 IF FLAG=1 AND INKEY(69)=0 AND BA<20
THEN PRINT CHR$(7): GOSUB 640
620 IF FLAG=2 AND INKEY(36)=0 AND BA>21
THEN PRINT CHR$(7): GOSUB 640
630 RETURN
640 LET HIT=0
650 WHILE HIT=0
660 LET CV=CV-1
670 IF SCREEN(BA,CV)=1 THEN GOTO 730
680 LOCATE BA,CV+1:PRINT " "
690 LOCATE BA,CV:PRINT CHR$(149)
700 IF CV=6 THEN LET FLAG=3-FLAG:LOCATE
BA,CV:PRINT " ":LET HIT=1:LET CV=21:LET
SHOT(3-FLAG)=SHOT(3-FLAG)-1:GOSUB 880
710 WEND
720 RETURN
730 IF FLAG=2 THEN GOTO 810
740 LOCATE BA,CV+1:PRINT " "
750 ENV 1,1,0,50,14,-1,10
760 SOUND 1,0,0,15,1,0,31:SOUND 2,0,0,15
,1,0,15:SOUND 4,0,0,15,1,0,1
770 LOCATE BA,CV:PRINT CHR$(238)
780 FOR U=1 TO 100:NEXT
790 LOCATE BA,CV:PRINT " "
800 LET FLAG=2:LET HIT=1:LET SCORE(1)=SC

```

```

ORE(1)+1:LET SHOT(1)=SHOT(1)-1:GOSUB 880
:LET SCREEN(BA,CV)=0:CV=21:GOTO 710
810 LOCATE BA,CV+1:PRINT " "
820 ENV 1,1,0,50,14,-1,10
830 SOUND 1,0,0,15,1,0,31:SOUND 2,0,0,15
,1,0,15:SOUND 4,0,0,15,1,0,1
840 LOCATE BA,CV:PRINT CHR$(238)
850 FOR U=1 TO 100:NEXT
860 LOCATE BA,CV:PRINT " "
870 LET FLAG=1:LET HIT=1:LET SCORE(2)=SC
ORE(2)+1:LET SHOT(2)=SHOT(2)-1:GOSUB 880
:LET SCREEN(BA,CV)=0:CV=21:GOTO 710
880 REM SCREEN DISPLAY
890 LOCATE 1,1:PEN 3:PRINT "PLAYER 1 ";:
PEN 2:PRINT A$
900 LOCATE 21,1:PEN 3:PRINT"PLAYER 2 ";:
PEN 2:PRINT B$
910 LOCATE 1,2:PEN 3:PRINT "SHOTS LEFT "
;:PEN 2:PRINT SHOT(1)
920 LOCATE 21,2:PEN 3:PRINT "SHOTS LEFT
";:PEN 2:PRINT SHOT(2)
930 LOCATE 1,3:PEN 3:PRINT "SCORE ";:PEN
2:PRINT SCORE(1)
940 LOCATE 21,3:PEN 3:PRINT "SCORE ";:PE
N 2:PRINT SCORE(2)
950 LOCATE 1,4:PEN 3:PRINT "LEVEL ";:PEN
2:PRINT SP
960 LOCATE 21,4:PEN 3:PRINT "LEVEL ";:PE
N 2:PRINT SP
970 RETURN

```

DUCKSHOOT

In this game you are walking through the reeds with your gun, intent on shooting as many ducks as possible. The birds fly across the screen at random heights and when hit, fall gently to the ground. Your dog then collects the dead bird. You score five points for each hit, and when you have shot five birds they are taken off to the market and you receive 15 bonus points.

At the end of your time (set at 200, but you could change this if you like) the total score is shown and you are invited to have another game. You can use a joystick, or key 1 on the main keyboard to move the hunter to the left, 0 moves him to the right, and press SPACE to fire the gun.

Detailed description

30-240 Install the user-defined graphics characters.

250-260 Set the ink, paper and border colours.

270-310 Clear the screen then print the instructions for playing the game.

320-330 Clear the screen and set the time to zero.

340-360 Print the screen display.

370-390 Initialize the variables.

400-410 Print the hunter on the screen.

420 Start of the loop controlling movement.

430 Update and print time and score.

440 Check that time has not run out.

450-460 Routine for controlling the wing movement of the duck.

470 Erase the duck.

480 Print the duck in its current position.

490 Check for bullet in motion.

500-520 Scan the keyboard for entries. Generate the appropriate sound for hunter moving or firing.

530-540 Print the hunter in the new position.

550 Go to sub-routine at line 660 if a bullet is in motion.

560 End of the movement loop.
 570 Erase the duck.
 580 Return to start of movement loop.
 600-650 Print the hunter in motion.
 660-720 Bullet firing routine. Print the hunter in the firing position and print bullet in motion.
 730-960 Print the dog along the bottom line moving to the dead duck, then returning to the left-hand side of the screen.
 970-1010 Print the dead duck at the bottom of the screen and increment the score.
 1020 Clear the buffer.
 1030 End of game routine. Print the score and invite the player to have another game.

```

10 REM DUCK SHOOT
20 REM T.BANKS
30 SYMBOL AFTER 234
40 SYMBOL 235,16,8,138,76,168,106,28,255

50 SYMBOL 236,0,26,193,38,24,193,38,153
60 SYMBOL 237,0,255,0,255,0,0,255,0
70 SYMBOL 238,1,49,49,1,127,251,237,153
80 SYMBOL 239,120,120,72,72,72,72,72,206
90 SYMBOL 240,0,48,48,0,120,252,250,186
100 SYMBOL 241,56,62,34,34,35,32,32,56
110 SYMBOL 242,120,56,40,40,232,136,136,
14
120 SYMBOL 243,0,12,12,0,30,63,93,93
130 SYMBOL 244,28,124,68,68,196,4,4,28
140 SYMBOL 245,30,28,20,20,23,17,17,112
150 SYMBOL 246,0,24,62,224,240,240,224,1
92
160 SYMBOL 247,64,64,96,63,63,31,15,3
170 SYMBOL 248,1,3,3,27,252,15,0,0
180 SYMBOL 249,128,128,128,220,127,192,0
,0
190 SYMBOL 250,0,0,7,30,253,11,3,6
200 SYMBOL 251,0,0,128,92,255,128,0,0
210 SYMBOL 252,98,225,63,62,34,34,68,0
220 SYMBOL 253,70,135,252,124,68,68,34,0
230 SYMBOL 254,97,225,63,62,66,66,33,0
240 SYMBOL 255,134,135,252,124,66,66,132
,0
  
```

```

250 INK 0,15:INK 1,9:INK 2,0:INK 3,26
260 PAPER 3:CLS:BORDER 26:PEN 1
270 CLS:LOCATE 15,1:PRINT "DUCKSHOOT":PR
INT:PRINT"THIS IS A GAME OF SKILL AT DUC
KSHOOTING.":PRINT"YOU ARE STOMPING THROU
GH THE MARSH LAND WAITING FOR POOR UNSUS
PECTING DUCKS TO FLY OVER."
280 PRINT:PRINT"EVERY DUCK YOU SHOOT IS
LAID OUT ON THE GROUND AROUND YOU, SO YO
U MAY COUNT YOURSPOILS. WHEN YOU HAVE SH
OT 5 DUCKS, THEYARE TAKEN OFF TO MARKET,
AND YOU ARE GIVEN 15 BONUS POINTS"
285 PRINT:PRINT "YOU HAVE A TIME LIMIT O
F 200 TO SHOOT ASMANY DUCKS AS YOU CAN."
290 PRINT:PRINT"CONTROLS ":"PRINT:PRINT"
KEYBOARD : 1-LEFT
          0-RIGHT
          <SPACE>-FIRE ":PRINT"OR USE A
JOYSTICK"
300 LOCATE 5,25:PRINT "PRESS ANY KEY TO
START"
310 IF INKEY$="" THEN GOTO 310
320 CLS
330 DATUM=INT(TIME/300)
340 LOCATE 1,24:PEN 2:PRINT STRING$(40,C
HR$(237))
350 LOCATE 1,23:PEN 0:PRINT STRING$(40,C
HR$(236))
360 LOCATE 1,22:PEN 1:PRINT STRING$(40,C
HR$(235))
370 LET X=2:LET BUL=20
380 LET DU=INT(RND*17)+2
390 LET A$=CHR$(240):LET B$=CHR$(239)
400 LOCATE X,21:PEN 2:PRINT A$
410 LOCATE X,22:PEN 2:PRINT B$
420 WHILE C<39
430 LOCATE 1,1:T1=INT((TIME/300)-DATUM):
PEN 2:PRINT "TIME ";T1:LOCATE 20,1:PRINT
"SCORE ";SCORE
440 IF T1>200 THEN GOTO 1020
450 LET C=C+1
460 IF C=1 THEN GOTO 480
470 LOCATE C-1,DU:PRINT " "
480 LOCATE C,DU:PEN 1:IF INT(C/2)=C/2 TH
EN PRINT CHR$(250)+CHR$(251):ELSE PRINT

```

```

CHR$(248)+CHR$(249)
490 IF FLAG=1 THEN GOTO 530
500 IF (JOY(0)=4 OR INKEY(64)=0) AND X>1
    THEN LET X=X-1:SOUND 2,25,5,7,0,0,1:GOT
    O 600
510 IF (JOY(0)=8 OR INKEY(32)=0) AND X<4
    THEN LET X=X+1:SOUND 2,25,5,7,0,0,1:GO
    TO 630
520 IF JOY(0)=16 OR INKEY(47)=0 THEN LET
    FLAG=1:SOUND 1,0,25,14,0,0,2
530 LOCATE X,21:PEN 2:PRINT A$
540 LOCATE X,22:PEN 2:PRINT B$
550 IF FLAG=1 THEN GOSUB 660
560 WEND
570 LOCATE C-1,DU:PRINT "   "
580 LET C=0:LET DU=INT(RND*18)+2
590 GOTO 420
600 LOCATE X+1,21:PRINT " ":LOCATE X+1,2
    2:PEN 1:PRINT CHR$(235)
610 LOCATE X,21:PEN 2:IF X/2 = INT(X/2)
    THEN PRINT CHR$(243):LOCATE X,22:PRINT C
    HR$(244):ELSE PRINT CHR$(243):LOCATE X,2
    2:PRINT CHR$(245)
620 GOTO 550
630 LOCATE X-1,21:PRINT " ":LOCATE X-1,2
    2:PEN 1:PRINT CHR$(235)
640 LOCATE X,21:PEN 2:IF X/2 = INT(X/2)
    THEN PRINT CHR$(243):LOCATE X,22:PRINT C
    HR$(241):ELSE PRINT CHR$(243):LOCATE X,2
    2:PRINT CHR$(242)
650 GOTO 550
660 LET BUL=BUL-1
670 LOCATE X,21:PRINT CHR$(238)
680 LOCATE X,BUL+1:PRINT "   "
690 LOCATE X,BUL:PEN 0:PRINT CHR$(148)
700 IF C=X AND BUL=DU OR X=C+1 AND BUL=D
    U THEN GOTO 730
710 IF BUL=2 THEN LET FLAG=0:LOCATE X,BU
    L:PRINT " ":LET BUL=20:RETURN
720 RETURN
730 LOCATE C,DU:PRINT "   "
740 FOR Z=DU TO 22
750 LET C=C+1
760 LOCATE C-1,Z-1:PRINT "   "

```

```

770 IF C=38 THEN LET C=1
780 LOCATE C,Z:PEN 1:PRINT CHR$(250)+CHR
$(251):SOUND 4,25,12,4
790 FOR I=1 TO 100:NEXT
800 NEXT
810 FOR Q=1 TO C
820 IF Q=1 THEN GOTO 850
830 IF Q-1=X THEN GOTO 850
840 LOCATE Q-1,22:PEN 1:PRINT CHR$(235)
850 LOCATE Q,22:PEN 0:IF INT(Q/2)=Q/2 TH
EN PRINT CHR$(253):ELSE PRINT CHR$(255)
860 IF Q=X THEN PEN 2:LOCATE Q,22:PRINT
CHR$(239)
870 FOR U=1 TO 100:NEXT
880 NEXT
890 LOCATE C+1,22:PEN 1:PRINT CHR$(235)
900 FOR E=C TO 1 STEP -1
910 IF E+1=X THEN GOTO 930
920 LOCATE E+1,22:PEN 1:PRINT CHR$(235)

930 LOCATE E,22:PEN 0:IF INT(E/2)=E/2 TH
EN PRINT CHR$(252):ELSE PRINT CHR$(254)
940 IF E=X THEN PEN 2:LOCATE E,22:PRINT
CHR$(239)
950 FOR J=1 TO 100:NEXT
960 NEXT
970 LOCATE 1,22:PEN 1:PRINT CHR$(235)
980 LET COUNT=COUNT+1:LET SCORE=SCORE+5
990 LOCATE COUNT*2,25:PEN 2:PRINT CHR$(2
47)+CHR$(246)
1000 IF COUNT=6 THEN LET SCORE=SCORE+15:
LOCATE 4,25:LET COUNT=1:PRINT "
"
1010 LET C=0:FLAG=0:LET BUL=20:LET DU=IN
T(RND*18)+2:GOTO 420
1020 FOR K=1 TO 255:LET J$=INKEY$:NEXT
1030 CLS:LOCATE 15,1:PRINT "DUCKSHOOT":P
RINT:PRINT"YOU OBTAINED A SCORE OF":PRIN
T:PRINT SCORE
1040 LOCATE 5,25:PRINT "PRESS ANY KEY TO
PLAY AGAIN"
1050 IF INKEY$="" THEN GOTO 1050
1060 RUN 320

```


VIRUS

In this game your objective is to destroy the virus which rushes about the screen under computer control. You do this by trapping it with the growing antibody which is under your control. This is a speedy game and you also have to be very careful not to crash into the walls. If you are successful in destroying the first virus you are presented with two to trap, and so on. You can move the antibody either with a joystick or the following keys:

QUp
ZDown
ILeft
PRight

Detailed description

10-30 Install the user-defined graphics characters.
40-130 Clear the screen and print the instructions for playing the game.
140 Dimension the screen array.
150-160 Dimension the variables.
170-200 Set the variable array to zero.
210-270 Print the border round the playing area.
280-310 Initialize the variables for the movements across the screen.
320-330 Work out a random position for the virus.
340 Beginning of the loop to print the virus.
350 Check that the virus has not run into an obstruction.
360-370 Calculate new co-ordinates for the virus.
380 Check to see if the new co-ordinates for the virus will meet an obstruction and if so go to 430.
390 Print the virus.
400 End of movement loop.
410 Go to the sub-routine at line 540 to scan the keyboard.
420 Loop back to line 340 to print the virus.

430-530 Determine if there are any vacant co-ordinates for the virus.
540-570 Scan the keyboard for entries.
580 Update the co-ordinates of the antibody.
590 Check to ascertain if the antibody has run into an obstruction.
600 Print the antibody.
610 Return for the next cycle.
620-660 End of game routine. Print a message and invite the player to play again.

```
10 SYMBOL AFTER 240
20 SYMBOL 255,&81,&7E,&66,&5A,&5A,&66,&7E,&81
30 SYMBOL 254,&E7,&C3,&A5,&18,&18,&A5,&C3,&E7
40 PEN 5:CLS:LOCATE 18,1:PRINT "VIRUS"
50 PRINT:PRINT"YOU CONTROL AN ANTIBODY THAT MUST TRAP AVIRUS BY MAKING THE VIRUS MOVE INTO A POSITION FROM WHICH IT CANNOT ESCAPE."
60 PRINT:PRINT"THE FIRST SCREEN CONTAINS ONE VIRUS, ANDIF YOU ARE SUCCESSFUL IN DESTROYING IT THE NEXT SCREEN CONTAINS 2, THE NEXT SCREEN 3, AND SO ON."
65 PEN 2:PRINT:PRINT "ANTIBODY"
- ";CHR$(254):PRINT:PRINT "VIRUS"
S - ";CHR$(255)
70 PEN 5:LOCATE 5,25:PRINT "PRESS ANY KEY TO CONTINUE"
80 IF INKEY$="" THEN GOTO 80
90 CLS:PRINT"CONTROLS : "
100 PRINT:PRINT"KEYBOARD : Q - UP
Z - DOWN
I - LEFT
P - RIGHT"
110 PRINT:PRINT"JOYSTICK CONTROL IS ALSO AVAILABLE"
120 LOCATE 5,25:PRINT "PRESS ANY KEY TO CONTINUE"
130 IF INKEY$="" THEN GOTO 130
140 DIM SCREEN(40,25)
```

```

150 DIM OPP(10,2):DIM XI(10):DIM YI(10):
LET COUNT=1
160 DIM VA(10)
170 FOR MP=1 TO 10:LET VA(MP)=0:NEXT
180 FOR IOP=1 TO 40
190 FOR FG =1 TO 25
200 LET SCREEN(IOP,FG)=0:NEXT:NEXT
210 LET A$=CHR$(143)
220 CLS: LOCATE 1,1
230 PRINT STRING$(28,A$)
240 LOCATE 1,24:PRINT STRING$(28,A$)
250 FOR A=2 TO 23:LOCATE 1,A:PRINT A$:LE
T SCREEN(1,A)=1:LET SCREEN(40,A)=1
260 NEXT
270 FOR B=2 TO 23:LOCATE 40,B:PRINT A$:N
EXT
280 FOR C=1 TO 40:LET SCREEN(C,1)=1:LET
SCREEN(C,24)=1:NEXT
290 A=10:B=3
300 LET IA=1:IB=0
310 LET Y=5
320 FOR L=1 TO 10:LET OPP(L,1)=INT(RND*2
0)+1:OPP(L,2)=INT(RND*20)+1:NEXT
330 FOR Q=1 TO 10:LET IX(Q)=1:NEXT
340 FOR K=1 TO COUNT
350 IF VA(K)=1 THEN GOTO 400
360 LET OPP(K,1)=OPP(K,1)+IX(K)
370 LET OPP(K,2)=OPP(K,2)+IY(K)
380 IF SCREEN(OPP(K,1),OPP(K,2))=1 THEN
GOTO 430
390 LOCATE OPP(K,1),OPP(K,2):PEN K:PRINT
CHR$(255):LET SCREEN(OPP(K,1),OPP(K,2))
=1:PEN 1
400 NEXT K
410 GOSUB 540
420 GOTO 340
430 IF IX(K)=1 THEN LET OPP(K,1)=OPP(K,1
)-1
440 IF IX(K)=-1 THEN LET OPP(K,1)=OPP(K,
1)+1
450 IF IY(K)=1 THEN LET OPP(K,2)=OPP(K,2
)-1
460 IF IY(K)=-1 THEN LET OPP(K,2)=OPP(K,
2)+1

```

```

470 IF SCREEN(OPP(K,1)+1,OPP(K,2))=0 THE
N LET IX(K)=1:IY(K)=0:GOTO 360
480 IF SCREEN(OPP(K,1)-1,OPP(K,2))=0 THE
N LET IX(K)=-1:IY(K)=0:GOTO 360
490 IF SCREEN(OPP(K,1),OPP(K,2)+1)=0 THE
N LET IX(K)=0:IY(K)=1:GOTO 360
500 IF SCREEN(OPP(K,1),OPP(K,2)-1)=0 THE
N LET IY(K)=-1:IX(K)=0:GOTO 360
510 LET VA(K)=1
520 FOR KL=1 TO COUNT:IF VA(KL)=1 THEN N
EXT:GOTO 530:ELSE GOTO 400
530 LET COUNT=COUNT+1:GOTO 170
540 IF JOY(0)=1 OR INKEY(67)=0 THEN LET
IB=-1:IA=0
550 IF JOY(0)=2 OR INKEY(71)=0 THEN LET
IB=1:IA=0
560 IF JOY(0)=4 OR INKEY(35)=0 THEN LET
IA=-1:IB=0
570 IF JOY(0)=8 OR INKEY(27)=0 THEN LET
IA=1:IB=0
580 LET A=A+IA:LET B=B+IB
590 IF SCREEN(A,B)=1 THEN GOTO 620
600 LOCATE A,B:PRINT CHR$(254):SCREEN(A,
B)=1
610 RETURN
620 FOR o=1 TO 255:LET jk$=INKEY$:NEXT
630 CLS:PRINT "BAD LUCK YOU CAUGHT THE V
IRUS AND DIED"
640 LOCATE 5,25:PRINT "PRESS ANY KEY TO
PLAY AGAIN"
650 IF INKEY$="" THEN GOTO 650
660 RUN 140

```

If you are a games addict continuously on the prowl for more and better games or looking for your first collection of listings to type in — look no further.

This sparkling collection will keep you busy for a long time. Some of the best programs that have ever appeared in the well-known magazine *Personal Computer World* have been converted for your computer and put together here in this prize collection.

Each game has been thoroughly tested and comes with full instructions, and a useful description of the programmer's technique to help you pick up some hints as you go.

Test your reflexes on arcade-style games or test your wits on strategy games — there is something here for everyone.

These games are also available on cassette tape to save you typing them all in.

ISBN 0-7126-0652-1

£5.95

ISBN 0 7126 0652 1

CENTURY COMMUNICATIONS LTD



9 780712 606523



DRY
THERM
STRIP
FOR
ZINC
COATING
ON
STEEL
AND
ALUMINUM

AMSTRAD CPC



MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA



<https://acpc.me/>

[FRA] Ce document a été préservé numériquement à des fins éducatives et d'études, et non commerciales.

[ENG] This document has been digitally preserved for educational and study purposes, not for commercial purposes.

[ESP] Este documento se ha conservado digitalmente con fines educativos y de estudio, no con fines comerciales.